


ひかりをあやつるLED（エルイーディー）

IchigoJamのスイッチを入れて、LED1（エル、イー、ディー、いち）とキーボードからうちこんで、Enter（エンター）キーをおしてみよう。（Enterキーは、みぎのようなおおきなキー）



LED1

「OK（オーケー）」とでて、IchigoJamのLEDがひかったら、だいせいこう！

LED0（エル、イー、ディー、ゼロ）、エンターでけせる。

LED0

キーボードのまんなかしたにあるなにもかいてない大きなキーはスペースキー。LED 1（エル、イー、ディー、スペース、いち）、エンターと、スペースはあってもなくてもOK。

LED 1

RED0（アール、イー、ディー、ゼロ）、エンターで、LEDはきえるかな？

RED0


「Syntax error（シンタックス・エラー）」とでて、きえません。なんとまちがえても、おこらないのがコンピューター。

ABD（エー、ビー、ディー）とうってみましょう。エンターキーはおさない。

ABD

ここで Backspace（バックスペース）キー をひとおし。

Backspaceキーは、キーボードの右上のほうにあります。



AB

ひとつもじけせました。これで、うちまちがいもこわくない。

やってみよう！

1. ABCDEFG とかいてみよう
2. LEDをけしてみよう
3. すばやくLEDをつけてけそう
4. BEEPとかいてエンターおしてみよう
5. CLSとかいてエンターおしてみよう

じかんをあやつるWAIT（ウエイト）

IchigoJamに「まて」をしてみよう。

がめんで、てんめつする、しかくい「カーソル」はキーボードがつかえるしるし。

```
WAIT60+
```

このカーソルがちょっとのあいだ、きえたの、わかりましたか？もういっかいためてみよう。

WAIT（ウエイト）のうしろにつくかずで、「まて」のながさをかえられます。

※1びょう=60

```
WAIT120+
```

カーソルがかえってくるまでのじかん、さっきよりながくなりました。

LEDをひからせて、じどうてきにけしましょう。

```
LED1:WAIT120:LED0+
```

コロン「:」（セミコロン「;」とまちがえやすい）をつかって、コマンドをれんけつ。LED1で光らせて、WAIT120でまって、LED0でけします。

チカッとひかるLED。

```
LED1:WAIT3:LED0+
```

WAITのかずをかえるとひかりかたがかわります。

LEDは2どひかる。

```
LED1:WAIT3:LED0:WAIT3:LED1:WAIT3:LED0+
```

コロン「:」でどんどんつなげられます。

やってみよう！

1. 10びょう「まて」をさせてみよう
2. LEDを5びょうつけて、けそう
3. WAIT10000とやって、とちゅうで[ESC]キーをおしてみよう
4. ながくひかったあと、チカッとひからせよう
5. カッコいいパターンでひからせよう

プログラムのきおく RUN/LIST

LEDのてんめつパターンをきおくしよう

```
10 LED1:WAIT10↵
```

かずではじめると「きおくしろ」というコマンド、まだLEDはひかりません
うごかすコマンドは「RUN (ラン)」エンター、LEDがひかります

```
RUN↵
```

つぎは20ではじめて、エンターで、きおくをついか

```
20 LED0:WAIT10↵
```

きおくさせたプログラムをうごかすときは、RUN
もういちど、RUNでひからせよう！

きおくをよびだす、LIST (リスト) コマンド

```
LIST↵  
10 LED1:WAIT10↵  
20 LED0:WAIT10↵
```

10ぎょうと20ぎょうのあいだに、WAITをついか

```
15 WAIT50↵
```

LISTでかくにん、RUNでじっこう！

ばんごうがおなじプログラムはうわがき (RUNとLISTでチェック！)

```
15 WAIT110↵
```

かずだけいれてエンターで、そのぎょうをけす

```
15↵  
LIST↵
```

やってみよう！

1. RUNのかわりファンクションキーF5キーをつかってみよう
2. LISTのかわりF4キーをつかってみよう
3. 2かいひかるプログラムをつくってみよう
4. かわいいパターンでひかるプログラムをつくろう！

プログラムのほぞん SAVE/LOAD/NEW

つくったLEDがひかるプログラム、ほぞんするときはSAVE（セーブ）コマンド

```
SAVE0↵  
Saved 30byte↵  
OK↵
```

ようりょうとOKができれば、ほぞんかんりょう！

IchigoJamほんたいに、0から3まで、4つほぞんできます

いちどでんげんをきって、LOAD（ロード）コマンドでよみこんでみましょう

```
LOAD0↵  
Loaded 30byte↵  
OK↵
```

LISTでかくにん！RUNでじっこう！

ほぞんしたプログラムをかくにんするFILES（ファイルズ）コマンド

```
FILES↵
```

まちがっておきにいのプログラムをけさないようにチェック

あたらしくプログラムをつくるときには、NEW（ニュー）コマンド

```
NEW↵  
OK↵  
LIST↵  
OK↵
```

LISTでなにもひょうじされなくなります

0ばんにほぞんしたプログラムは、ほんたいボタンをおしながらでんげんスイッチをいれると、じどうきどう（キーボードやテレビをはずしてもOK、こうさくなどにどうぞ！）

やってみよう！

1. 3かいひかるプログラムをほぞんしよう
2. じどうきどうをつかってみよう
3. NEWしてSAVEするとどうなるかチェックしよう

よるのしんごうをつくる GOTO (ゴートゥー)

よるのてんめつしんごう、0.5びょうごとについたりきえたり。

```
10 LED1:WAIT30:LED0:WAIT30↵  
20 LED1:WAIT30:LED0:WAIT30↵  
RUN↵
```

10ぎょうをうちこんだら、もどって、せんとうの10を20にかえてエンター
LISTで、20ぎょうにコピーされた10ぎょうをかくにん！

あさになるまで、やく3まんかい。ずっとてんめつさせる、GOTO (ゴートゥー) コマンド。

```
10 LED1:WAIT30:LED0:WAIT30↵  
20 GOTO 10↵
```

GOTOのうしろのかずは、ジャンプさき

> 10ぎょう、1かいてんめつ

> 20ぎょう、10ぎょうへジャンプ！

とめるときは、[esc] (エスケープ) キー

こうそくてんめつ

```
LIST↵  
10 LED1:WAIT3:LED0:WAIT3↵
```

LIST(=F4)でひょうじ

10ぎょうの、2つある30を3にかえて、エンター、RUN(=F5)

```
15 LED1:WAIT30:LED0:WAIT30↵
```

10ぎょうと20ぎょうのあいだに15ぎょうをつくろう

こうごにくりかえす、みじかいてんめつと、ながいてんめつ

やってみよう！

1. よるのしんごうプログラムにもどしてみよう
2. ひかっているじかんだけ、ながくしてみよう
3. こうそくにてんめつさせて、ひかっているときにとめるゲームにしよう
4. ぎょうのばんごうを100から100きざみでつくってみよう
5. すきなてんめつパターンをつくってだれかにじまんしよう

おもいをつたえる BTN (ボタン)

IchigoJamのひだりにあるボタンでコンピューターにおもいをつたえよう。

```
?BTN<>+
```

ボタンをおさないでじっこうすると、0 とでる。

てんめつするしかく（カーソル）を、キーボードのみぎした、「うえ」キーを3
かいおして、ボタンをおしながらエンター。1 とでたらせいこう！

「GOTO」コマンドをつかってくりかえし。

```
10 ?BTN<>+  
20 GOTO 10+  
RUN+
```

ずらっと0がでたら、ボタンをおしたりはなしたりしてみよう。

とめるときは、[esc]（エスケープ）キー

ボタンをおしたらLEDがひかるプログラム

```
10 LED BTN<>+  
20 GOTO 10+
```

RUNコマンドのかわりにF5キーでもOK！

LEDコマンドは1のときひかって、0のとききえるので、ボタンにあわせてひかる
よ。[esc]キーでとまる。

ボタンをおしたら、こうそくてんめつ！

```
10 LED 1:WAIT 10-BTN<>*5+  
20 LED 0:WAIT 10-BTN<>*5+  
30 GOTO 10+
```

WAIT（ウエイト）はまてのコマンド。ボタンがおされていないと、WAIT 10。お
されていたらBTN()が1になって、それを「*5」で5ばいしたかずをひくので、
WAIT 5。GOTO（ゴートウー）で10ぎょうにもどってくりかえし。

やってみよう！

1. ボタンがおされたら、もっとはやくてんめつさせよう
2. ボタンがおされたら、ゆっくりてんめつさせよう
3. ボタンがおされたら、ずっとひかるようにかいぞうしよう
4. ボタンをおしていないときは、まったくひからないようにかいぞうしよう

はんだんさせる IF（イフ）

ボールがとんできたらよける。あながあったらジャンプ。

コンピューターにはんだんさせるIF（イフ）コマンド。

IFはえいごで「もし」といういみ。

```
10 IF BTN()=0 GOTO 10+  
RUN+
```

もし、ボタンがおされていなかったら10ぎょうをくりかえし

ボタンをおしたらプログラムはつぎにすすもうとして、しゅうりょう。

ボタンをおすごとにLEDをつけたりけしたりしてみよう

```
10 IF BTN()=0 GOTO 10+  
20 LED1+  
30 IF BTN()=0 GOTO 30+  
40 LED0+  
50 GOTO 10+
```

おや？ボタンをおしているあいだ、つきっぱなし？

コンピューターははやいので、ボタンをおしたらLEDをひかせて、すぐつぎのはんだんにいってしまいます。

すぐにつぎにいかないよう、WAIT（ウェイト）コマンドをついかする

```
10 IF BTN()=0 GOTO 10+  
20 LED1:WAIT30+  
30 IF BTN()=0 GOTO 30+  
40 LED0:WAIT30+  
50 GOTO 10+
```

やってみよう！

1. ボタンをおしっぱなしにしてみよう
2. WAIT30をWAIT60にかえて、ためしてみよう
3. WAIT30をWAIT5にかえて、ためしてみよう
4. IF BTN()=1にかえて、ためしてみよう
5. ボタンがおされたら、てんめつしはじめるプログラムをつくろう
6. LEDをたのしくあやつるプログラムをつくろう

おとをあやつるBEEP（ビープ） - IchigoJam

サウンダーのあしをIchigoJamのSOUNDとGNDにさして

```
BEEP➤
```

ピツとなったら、せいこう！

たかいおと

```
BEEP 5➤
```

ひくいおと

```
BEEP 20➤
```

ながいおと

```
BEEP 10,30➤
```

リズムをつくろう

```
10 BEEP 5:WAIT 30➤  
20 BEEP 20:WAIT 30➤  
30 GOTO 10➤  
RUN➤
```

WAIT（ウエイト）はまてのコマンド、かずをおおきくするとゆっくりリズム。
GOTO（ゴートウー）で10ぎょうにもどってくりかえし。[ESC]キーでていし。

IchigoJamほんたいのボタンをおしているあいだ、リズムをストップ！

「(」はシフトおしながら「8」、「)」はシフト「9」でうちこむよ

```
15 IF BTN<> GOTO 15➤
```

やってみよう！

1. いろんなおとをながしてみよう
2. とってもはやいリズムをつくってみよう
3. 「12 BEEP 20:WAIT 30」をくわえて、3びょうしにしてみよう
4. 4びょうしをつくってみよう
5. 15行のGOTO 15 を GOTO 10 にかえるとどうなるかな？
6. オリジナルのリズムをつくってみよう

おとがくをつくるPLAY (プレイ) - IchigoJam

サウンダーのあしをIchigoJamのSOUNDとGNDにさして

```
PLAY"CDE"⏏
```

ドレミとなりましたか？ダブルクォート「"」は、[SHIFT]おしながら[2]

ドレミファソラシのおんかいは、CDEFGABであらわします。

```
PLAY"CDEFGAB"⏏
```

おとのたかさオクターブはO（アルファベットのオー）とかずでかえます。

```
PLAY"O4 CDEFGAB O5 C"⏏
```

おんかいにかずをつけるとながさがかわります。

```
PLAY"CDE2"⏏
```

ドレミーのミは2ぶおんぷの「2」。なにもつけないと4ぶおんぷ、8ではんぶんのながさの8ぶおんぷ。1でぜんおんぷ。

はんおんあげるシャープはおんかいのうしろに「#」（シャープ）、はんおんさげるフラットは「-」（マイナス）。

```
PLAY"C C# F-"⏏
```

おやすみはR（アール）、R2とおんかいとおなじようにながさのしてもOK。

```
PLAY"CRC"⏏
```

テンポをかえるのはT（ティー）、なにもしていなければ4ぶおんぷが1ぶんに120かいのテンポ120

```
PLAY"T240 CDE2 CDE2 GEDCED2"⏏
```

やってみよう！

1. チューリップのうたをつくってみよう
2. おんがくがなっているときに、PLAYとだけうちこんでみよう
3. かえるのうたをつくってみよう
4. ともだちとりんしょうしてみよう！（おんがくをずらすがっそう）
5. おんがくのきょうかしょからおんがくをつくってみよう！

けいさんさせるPRINT（プリント） - IchigoJam

けいさんがとくいなコンピューターに、もんだいをだしてみよう！
していされたものをひょうじするPRINT（プリント）コマンドをつかいます。

```
PRINT 1+1+
```

エンターキーで「2」とすぐにこたえてくれます

```
PRINT 15-7+
```

ひきざんだってできます！

コンピューターでは、かけざんきごうをアスタリスク「*」であらわします。

```
PRINT 3*5+
```

わりざんはスラッシュ「/」（ひだりがぶんし、みぎがぶんぽ）

```
PRINT 10/2+
```

じつは、PRINTとうつかわりに「?」もつかえます。（しょうりゃくけい）

```
?10/3+
```

おや、でもけっかがおかしい？

しょうすうをつかわないIchigoJamではこれがせいかい。あまりは「%」でもとめます。

```
?10%3+
```

10わる3は3あまり1なので、「1」とひょうじされます。

さいごにおおきなかずをけいさんさせてみましょう。

```
?1000*100+
```

おや、ようすがへんですね。

じつは、IchigoJamがけいさんできるはんいが「32767いか、-32768いじょう」ときまっています。

やってみよう！

1. 100たす200をけいさんさせよう
2. 8かける7をけいさんさせよう
3. 100わる25をけいさんさせよう
4. 30を8でわったあまりをけいさんさせよう
5. 1おく（1まんの1まんばい）をけいさんさせてみよう

ばしよをあやつるLOCATE（ロケイト）

していしたばしよにひょうじさせてみよう！

ばしよをしているLOCATE（ロケイト）コマンドをつかいます。

```
LOCATE 16,12:PRINT "*"␣
```

エンターキーでまんなかあたりに「*」とひょうじされます。

これはいちばんひだりから16すすんだところ、うえから12すすんだところにひょうじされるといういみです。

ちがうばしよにひょうじさせてみよう。

```
LOCATE 25,5:PRINT "*"␣
```

LOCATEのかずをかえることでひょうじされるばしよがかわります。

じつは、LOCATEのしょうりゃくで「LC」とうつこともできます。

```
LC 16,14:PRINT "*"␣
```

ふたつのばしよにもじをひょうじすることもできます。

```
LC 16,12:PRINT "@" : LC 10,10:PRINT "*"␣
```

けいさんしたこたえをひょうじすることもできます。

```
LC 16,14:PRINT 2+3␣
```

がめんをきれいにするときにはF1キーをおすか、「CLS」コマンド。

```
CLS␣
```

やってみよう！

1. ひだりから20、うえから10すすんだところに「IchigoJam」をひょうじさせてみよう
2. ひだりから10、うえから17すすんだところと、ひだりから25、うえから18すすんだところに「@」とひょうじさせてみよう
3. ひだりから15、うえから24すすんだところに10-5のけいさんけっかをひょうじさせてみよう
4. すきなところにすきなもじをひょうじさせてみよう
5. すきなところにすきなけいさんけっかをひょうじさせてみよう

ヘリコプターをうごかそう！ MOVE

ヘリコプターをさゆうにうごかしてみよう！

まずはじまりのばしょをしていしてあげよう！

```
10 X=16
```

ヘリコプターがはじめにいるばしょを16にしていしよう。

ヘリコプターをそうさするためのじゅんびをしよう！

```
20 CLS
30 K=INKEY()
```

INKEY()でおされたキーをしることができるよ。

ヘリコプターをさゆうにうごかすキーをせっていしよう！

```
40 IF K=28 THEN X=X-1
50 IF K=29 THEN X=X+1
```

28がやじるしキーのひだり、29がやじるしキーのみぎのいみだよ。

キーにゆうりよくとヘリコプターそうさをつなげよう！(キャラクターコード
243、ALT+J)

```
60 LOCATE X,10: ?CHR$(0,243,0)
70 GOTO 30
```

RUNでじっこうしてみると、どうなるかな？

やってみよう！

1. ヘリコプターいがいのモノをうごかしてみよう！
2. じょうげにうごかしてみよう！
3. くふうしてじゆうじざいにうごかしてみよう！

参考:IchigoJamで簡単なゲームを作ってみよう～お父さんが教えるプログラミング実践編⑥
<https://edtechzine.jp/article/detail/675>

CC BY IchigoJamプリント <http://ichigojam.net/print/> IchigoJam@jig.jp

サイコロをつくるRND（ランダム）

いろんなサイコロをつくってみよう！

?RND(3)+

どんなすうじがでたかな？

RND（ランダム）は（）（カッコ）のなかのかずだけめんがあるサイコロです。

もういちどうって、エンターキーをおしてみよう！

?RND(3)+

ひょうじされたすうじがさいしょとかわったかな？

あれ？でも3がでない？

これは0をあわせた3つのすうじがでるといういみです。（）（カッコ）のなかか3のばあいには0、1、2の3つになります。

?RND(5)+

（）（カッコ）のなかのすうじをかえることもできます。

0をだしたくない！でてくるかずでいちばんちいさいかずを1にしたい！

そんなときは...

?RND(3)+1+

これでだいじょうぶ！

いちばんちいさいかずだった0に1をたして、でてくるかずでいちばんちいさいかずが1になります。

?RND(3)+3+

でてくるかずのなかでいちばんちいさいかずをかえることもできます。

やってみよう！

1. 0から6のすうじのどれかをランダムにひょうじさせてみよう
2. 1から6のすうじのどれかをランダムにひょうじさせてみよう
3. 10から20のすうじのどれかをランダムにひょうじさせてみよう
4. すきなかずがでるサイコロをつくってみよう

ストップウォッチをつくる TICK (ティック)

1、2、3、4、5、6・・・。

1びょうに60かい、じかんをかぞえつづけるIchigoJam。

```
?TICK()↵
```

TICK (ティック) は、そのかず。なんどかひょうじさせてみよう。

```
?TICK()↵
```

2かいめにひょうじされるかずは、1かいめよりおおきくなる。

```
?TICK()/60↵
```

60でわることで、でんげんをいれてからのじかんを、びょうにかえる。

かぞえているかずをリセットするコマンドCLT (クリアタイマー)

```
CLT↵  
OK↵  
?TICK()/60↵
```

リセットしてからのじかんがわかる。

RUNしてから、ボタンをおすまでのじかんをびょうではかるストップウォッチ

```
10 CLT↵  
20 IF BTN()=0 GOTO 20↵  
30 ?TICK()/60↵
```

20ぎょうでボタンがおされるまで、くりかえしてまっている

ボタンをおすたびにじかんをひょうじするストップウォッチに。

```
40 IF BTN()=1 GOTO 40↵  
50 GOTO 10↵
```

やってみよう！

1. いきをおおきくすって、なんびょういきをはきつづけられるかはかるう
2. めをつむって、5びょうかんかくになるようにボタンをおしてみよう
3. 30ぎょうの/60をけして、じぶんのボタンれんだそくどをはかるう
4. SAVE0でほぞん、でんげんをきって、ボタンをおしながらでんげんをいれて、じどうきどうさせたら、れんだしてみよう！
5. ボタンをおしてかいし、2どめのボタンでひょうじするようかいぞうしよう

かずをきおくするLET (レット) - IchigoJam

きおくりよくばつぐんのコンピューター、へんすうA~Zにかずをきおくできる！

```
LET A,1190:LET B,2014:LET C,794
```

OKとでたら、きおくかんりょう！

がめんをけして、Aは？Bは？Cは？

```
CLS
?A
?B
?C
```

もういちどLETで、きおくをうわがき

```
LET B,2017
?B
```

かずには、へんすうもつかえます

```
LET C,A+B
?C
LET C,C+1
?C
```

「= (イコール) きごうをつかって、みじかくかくこともできます。

```
C=C+1
```

だんだんひくくなるおと！

```
10 LET A,1
20 ?A:BEEP A:WAIT 10:LET A,A+1
30 IF A=10 END
40 GOTO 20
```

やってみよう！

1. もっとひくくなるおとをつくろう
2. だんだんたかくなるおとをつくろう
3. うまれたとし、つき、ひをおぼえさせ、ぜんぶたしたかずをきおくさせよう
4. うまれたつきとひをかけざんしたかずを、きおくさせて、ひょうじさせよう
5. なんけたまできおくできるかためしてみよう

かずでこたえる INPUT

RND（ランダム）をつかって、いろんなたしざんもんだいをだすプログラム
A～ZにかずをきおくするLET（レット）をつかうよ

```
10 LET A,RND(10)+  
20 LET B,RND(10)+  
30 ?A;"+";B+  
RUN+
```

なんか、RUN（ラン）してみましょう！

キーボードをつかってかずをきおくさせるにはINPUT（インプット）

```
40 INPUT C+  
50 ?C+  
RUN+
```

ひょうじされた「?」のあとに、キーボードでかずをいれて、エンターキー
Cに、いれたかずが、きおくされているね！

こたえあわせには、はんだんさせるIF（イフ）

```
60 IF C=A+B ?"ATARI"+
```

はずれをひょうじしたいときは、ELSE（エルス）をつかおう

```
60 IF C=A+B ?"ATARI" ELSE ?"HAZU"  
RE"+
```

しつもんぶんつきで、きくこともできるよ

```
40 INPUT "KOTAEHA?",C+
```

やってみよう！

1. ふたけたのたしざんに、かいぞうしてみよう
2. ちょうかんたんなもんだいに、かいぞうしてみよう
3. かけざんもんだいに、かいぞうしてみよう
4. TICKをつかって、こたえるまでのじかんをはかってみよう
5. かずあてゲームをつくってみよう

くりかえしでかけざん FOR/NEXT (フォー・ネクスト) - IchigoJam

```
10 A=5↵
20 FOR I=1 TO A:?"*";:NEXT:↵
```

FOR/NEXT (フォー・ネクスト) コマンドで、へんすうIを1から1ずつふやしながら、Aになるまでくりかえし。*マークが5つでました。

INPUT (インプット) コマンドをつかって、かずをにゅうりょくできるようにしよう！

```
10 INPUT A↵
```

いろいろかずをいれてみよう

もうひとつかずをにゅうりょくできるようにして、そのかずのぶんだけ、さらにくりかえして！

```
15 INPUT B↵
17 FOR J=1 TO B↵
30 NEXT↵
```

かけざんができました！

へんすうCをよういして、*マークのかずをかぞえてもらいましょう

```
16 C=0↵
25 C=C+A↵
40 ?C↵
```

くりかえすごとに、*マークがAこずつふえるので、 $C=C+A$ 。LET C,C+1 としてもOKです。C=0 または LET C,0 としておかないと、じっこうするたびにCのかずがどんどんふえてしまいます！

やってみよう！

1. 7かける8をけいさんさせてみよう
2. 8かける7をけいさんさせてみよう
3. マークをすきなマークに変えてみよう
4. マークを2もじにするとどうなるでしょう？
5. 0をにゅうりょくするとどうなるかじっけんしよう

2しんすうをつくるBIN\$ (バイナリー)

すうじ、ひらがな、アルファベット、いろんなもじがありますが、コンピューターがつかうもじは「0」と「1」の2つだけ。

ふつう、9に1たすと、10と2もじになりますが、0と1しかないので、1に1をたしただけで、10（いちぜろ）と2もじになります。これが2しんすう。

かずを2しんすうにかえるBIN\$ (バイナリー) コマンド

```
?BIN$(2)↵  
10  
?BIN$(10)↵  
1010
```

2しんすうをかずにする、きごう「」(shift+@か1のひだり)

```
?`1111↵  
15  
?`1000000000↵  
256
```

0,1,2,3,4...2しんすうでかぞえさせてみます。

```
FOR I=0 TO 256: ?BIN$(I), I: WAIT 3  
0: NEXT↵  
0 0  
1 1  
10 2  
11 3
```

2しんすうは、bit (ビット) でかぞえます。3は11で2bit。8は1000で4bit。

やってみよう！

1. 16を2しんすうにしてみよう
2. 2しんすうの101、1010、10100をかずにしてみよう
3. 40を2しんすうにしてみよう
4. 5bitであらわせるさいだいのかずは？
5. -1を2しんすうにしてみよう

16しんすうをつくるHEX\$ (ヘックス)

コンピューターがつかうもじは「0」と「1」の2つだけ(2しんすう)。でもにんげんにはよみにくい。そこで16しんすうのではん

16しんすうでつかうすうじは0,1,2,3,4,5,6,7,8,9,A,B,C,D,E,F

ふつう、9に1たすと、10と2もじになりますが、16しんすうでは9のつぎはAになります。これが16しんすう。

かずを16しんすうにかえるHEX\$ (ヘックス) コマンド

```
?HEX$(2)↵
2
?HEX$(10)↵
A
```

16しんすうをかずにする、きごう「#」(shift+3)

```
?#16↵
10
?#100↵
256
```

0,1,2,3,4...16しんすうでかぞえさせていただきます。

```
FOR I=0 TO 256: ?HEX$(I), I: WAIT 3
0: NEXT↵
0 0
1 1
. . .
9 9
A 10
. . .
```

16しんすうではFのつぎは10(いちゼロ)になります

やってみよう！

1. 12を16しんすうにしてみよう
2. 16しんすうの1A、6F、ABCをかずに見てみよう
3. 40を16しんすうにしてみよう
4. 12けたであらわせるさいだいのかずは？
5. -1を16しんすうにしてみよう

オリジナルキャラをつくるPCG - IchigoJam

キーボードのひだりAltをおしながら、Cをおすとでる「ねこ」

```
?CHR$(236)␣
```

「ねこ」のキャラクターコードは、236

この「ねこ」を「ぼう」にかえてみせます

```
POKE 236*8,8,8,8,8,8,8,8,8␣
```

かわった!?

こんどはちいさいボールにかえてみせます

```
POKE 236*8,0,0,0,24,24,0,0,0␣
```

ななめのせんに!

```
POKE 236*8,1,2,4,8,16,32,64,128␣
```

2しんすう、0と1をつかって、すきなえをかくプログラム

```
10 POKE236*8+0,`01111110␣
11 POKE236*8+1,`11011011␣
12 POKE236*8+2,`11111111␣
13 POKE236*8+3,`11000011␣
14 POKE236*8+4,`11111111␣
15 POKE236*8+5,`11111111␣
16 POKE236*8+6,`00011000␣
17 POKE236*8+7,`00011000␣
RUN␣
```

「アイスくん」がとうじょう!

ねこにもどしたいときは CLP (クリアパターン)

```
CLP␣
```

やってみよう!

1. ハートマークをつくってみよう
2. すきなキャラをつくってみよう
3. 236を237にして、ALT+Dのうちゅうじんをかきかえよう(224-255までOK!)
4. ?`10000000 の2しんすうをひょうじしてみよう
5. オリジナルキャラをプログラム1ぎょうでつくってみよう

1たす1は1、ろんりのなぞ - IchigoJam

0と1しかない「ろんり」のせかいでけいさんしよう。まずはたしざん「|」

```
?1|0+
1+
?0|0+
0+
?1|1+
1+
```

くりあがりはないので、1に1をたしても1のまま。これが「ろんりわ」

つぎは「&」かけざん（コンマでくぎって、まとめてけいさん）

```
?1&1,0&1,1&0,0&0+
1      0      0      0+
```

1をかけたらもとのかずのまま。0をかけると0になる。これが「ろんりせき」

りょうほう1のときだけ0になる、かわったたしざん「^」

```
?1^1,0^1,1^0,0^0+
0      1      1      0+
?1|1,0|1,1|0,0|0+
1      1      1      0+
```

1^1のほかは「ろんりわ」といっしょ。これが「はいたてきろんりわ/XOR」

「ろんり」のけいさんは、さいだい16コマとめてけいさんできちゃいます！

```
?BIN$(1),BIN$(2),BIN$(3)
1011+
```

4けたの2しんすうどうしのたしざん。1|0、0|0、1|1、0|1がいちどにけいさんできました！

やってみよう！

1. 3|1はいくつでしょう？
2. ?BIN\$(1),BIN\$(2),BIN\$(3) をヒントに、3|1のなぞをとこう
3. ?8323^35 と ?8323^35^35 をけいさんしてみよう
4. いろんなかずにかえて、ろんりをつかったあんごうであそぼう
5. いろんなけいさんをさせて、ろんりのなぞにせまろう

プログラムをつなげる LRUN - IchigoJam

4つまでほぞんできるIchigoJam。プログラムどうしをつなげて、たいさくに！
まずはゲームほんぺん、クイズゲームをつくります

```
10 CLS#  
20 LC10,10:INPUT "1+1=?",N#  
30 LC12,10:IF N=2 ?"CLEAR!!" ELS  
E ?"GAME OVER!"#
```

せいかいでクリア！そうでなければゲームオーバーになります
これをファイル1にほぞんしておきます

```
SAVE1#
```

タイトルプログラムをつくるために、まずはさいしょから

```
NEW#
```

なにかキーがおされたらおわるタイトルプログラム

```
10 CLS#  
20 LC6,7:?"=MAGICAL MATH QUIZ="#  
30 WAIT 60#  
40 LC6,18:?"HIT ANY KEY TO START  
"#  
50 IF INKEY()=0 CONT#
```

れんけつコマンドLRUNをついかします

```
60 LRUN1#  
SAVE0#  
RUN#
```

タイトルプログラムからゲームへつながりました！

やってみよう！

1. LRUN0 でリトライ、ゲームオーバーにしてみよう
2. タイトルをかいぞうしてみよう
3. クイズをかいぞうしてみよう
4. ゲームがおわったらタイトルがめんにもどるようにしてみよう
5. かわくだりゲームにタイトルをつけてみよう

もじのしょうたい CHR\$/ASC

ASCをつかって、もじのしょうたいをあばいてみよう！

```
?ASC("A")
65
?ASC("B")
66
?ASC("C")
67
```

Aは65、Bは66、Cは67、1ずつちがう

CHR\$をつかうと、かずからもじにぎやくへんかん

```
?CHR$(65)
A
?CHR$(66)
B
?CHR$(67)
C
```

もじにたいおうするかずを「もじコード」をいうよ

FORコマンドでアルファベットのおおもじをならべよう

```
FOR I=65 TO 90: ?CHR$(I);:NEXT: ?
ABCDEFGHIJKLMNOPQRSTUVWXYZ
```

32をたすともじがならぶよ

```
FOR I=65 TO 90: ?CHR$(I+32);:NEXT: ?
abcdefghijklmnopqrstuvwxyz
```

えもじのもじコードをしらべたい？

プログラムですぐかいけつ！

```
FOR I=224 TO 255: ?CHR$(I);I;:NEXT
```

やってみよう！

1. きごう「!」のもじコードをしらべよう
2. ?CHR\$(34) でひょうじされるきごうをチェックしよう
3. いろんなカタカナのもじコードをしらべよう
4. アからンまでひょうじするプログラムをかいてみよう
5. ?CHR\$(511) でひょうじされるきごうをチェックしよう

がめんにいたずらしてみよう！ DRAW

がめんにせんをかいてみよう！ DRAWコマンドでせんをかくことができるよ

```
DRAW 1,1,70,70
```

2つばしょをしていしてあげよう

さいしょの1,1でさいしょのばしょ。

つぎの70,70,でつぎのばしょをしていしよう。

ばしょのつぎに0とくわえると、せんをけせるよ。

```
DRAW 1,1,70,70,0
```

うえのほうのもじがきえたかな？

FORコマンドでくりかえしせんをかこう！

```
10 FOR I=1TO11 STEP 3:DRAW I,1,I*10,70:NEXT
```

RUNでじっこうしてみると、どうなるかな？

FORコマンドをふたつかいて#をかいてみよう！

```
10 FOR I=1TO10 STEP 3:DRAW I*10,1,10*I,70:NEXT  
20 FOR I=2TO8 STEP 3:DRAW 1,I*5,70,I*5:NEXT
```

やってみよう！

1. STEPのかずをもっとちいさくしてみるとどうなるかな？
2. DRAWのI*のよこのかずをかえてみよう
3. くふうしてもじやえをかいてみよう！

CC BY IchigoJamプリント <http://ichigojam.net/print/IchigoJam@jig.jp>

はすのはなをえがこう！

DRAWコマンドでせんをひいて、はなをかいてみよう！

```
5  CLS : X=1 : A=0 : B=48
```

まずはじゅんぴから。

じっこうするたびにがめんをきれいにして、ついでにへんすうもせっていしよう。

```
10  A=A+5  
15  B=B-X
```

Aのなかのすうじがすこしずつふえるようにしよう。

Bはすこしずつへるようにしよう。

Bのおおきをちょうせつしよう

```
20  IF B>48 X=1  
25  IF B<10 X=-1
```

Bがおおくなりすぎたり、ちいさくなりすぎないように、Xをつかってふやすかへらすかをちょうせつしよう。

いよいよDRAWのでばん！

```
30  DRAW 32+COS(A)/B,24+SIN(A)/B  
35  GOTO 10
```

コサインやサインをつかうとまるをつくることができるんだ。

じゅんぴしたAやBをつかって、カーブをちょうせいしてはなびらをつくるよ！
じっこうしてみよう！

やってみよう！

1. 10ぎょうで、たすすうじをいじって、いろんなかたちのはなをつくってみよう
2. 20,25ぎょうのすうじをかえてみるとなにおこるかな？
3. 30ぎょうの32や24をかえてみるとなにおこるかな？

1から100までたす - IchigoJam

ハテナきごうをつかって、コンピューターに「1+1」を、きいてみよう？

```
?1+1↵
```

エンターキーで「2」とこたえてくれました。では「1+2」は？

```
?1+2↵
```

ボーリングのピンは、ぜんぶでなんほんあるでしょう？

```
?1+2+3+4↵
```

10ほんでした。1れつふやすとどうなるか？カーソルキーのうえとさゆうをつかって、プログラムをリサイクル！「+5」とつかでうちこんだら、エンターキー。

```
?1+2+3+4+5↵
```

このちょうしで1から10までたしてみるとこたえはいくつ？

```
?1+2+3+4+5+6+7+8+9+10↵
```

1から100までだと、たいへんそうですね。そこで、くりかえしのFORコマンド！

```
FOR I=1 TO 100: ?I: NEXT↵
```

へんすうAをよういして、どんどんたしてもらいましょう！

```
A=0↵
OK↵
FOR I=1 TO 100:A=A+I: NEXT↵
OK↵
?A↵
5050↵
```

やってみよう！

1. 1から12までたすといくつ？
2. 1から20までたすといくつ？
3. 1から200までたすといくつ？
4. 1から300までたすといくつ？
5. ちゃんとけいさんできるのはいくつまで？

かけざん名人 - IchigoJam

IchigoJamに九九を見てもらおう！

6x7 は、記号「*」を使って計算してくれます。

```
?6*7  
42
```

1～9、9種類のでたらめな数はRND(ランダム)を使います。

```
?RND(9)+1
```

それを2つ使って、九九の問題を出してもらおう（xはシフト押しながらX）

```
10 A=RND(9)+1:B=RND(9)+1  
20 ?A;"×";B;"=";  
RUN
```

RUNするたびに、いろんな問題が出てきます。

INPUTを使って、答えを入力できるようにします。

```
30 INPUT C  
RUN
```

IFを使って間違っていたら "BAD!" と出して終了です。

```
40 IF A*B!=C ?"BAD!":END
```

10回分のタイムを測ってみましょう。

```
5 CLT:FOR I=1 TO 10  
50 NEXT:?TICK()/60
```

何秒まで縮められるかな！？

やってみよう！

1. 問題を5問に減らしてみよう
2. 問題を20問に増やしてみよう
3. 1の段は出さないようにしよう
4. 19の段まで出すインド式にチャレンジしよう

じかんわりづくりとあまりのけいさん（さんすう）

わりざんさせよう！コンマをつかって、3つまとめてけいさん！

```
? 2 / 1 , 10 / 5 , 13 / 3 #  
1   2   4 #
```

おや？13わる3は4かな？たしかめてみよう

```
? 1 * 2 , 2 * 5 , 4 * 3 #  
2   10  12 #
```

4かける3は12、13にはならないね

IchigoJamのわりざん (/) は、しょうすうをきりすてるよ

「あまり」をけいさんしよう

13わる3は、4あまり1。このあまりをけいさんするには？

```
? 13 - (13 / 3) * 3 , 13 % 3 #  
1   1 #
```

IchigoJamのあまりをもとめるきごう (%) をつかうとかんたん！

1じかんは60ふん、0:00から90ふんたつとなんじ、なんぶん？

```
? 90 / 60 , 90 % 60 #  
1   30 #
```

1:30、とけいさんできるよ！

じぶんだけのじかんわりをつくってみよう！8:50スタート

```
10  H=8:M=50 #  
20  INPUT  T #  
30  M=M+T #  
40  ?H+M/60;" : " ; M%60 #  
50  GOTO  20 #
```

さいしょのじゅぎょうが45ふん、45といれてエンター。

おやすみが10ふん、10といれてエンター。どんどんじこくがひょうじされる！

やってみよう！

1. やすみじかんなしだと、なんじにがっこうおわる？
2. 1つのじゅぎょうが30ふんだとすると？
3. あさはやいじかんからはじめてみよう
4. どうようびのじかんわりをつくってみよう

くりかえしでわりざん - IchigoJam

12コのイチゴを、3にんでわけてみましょう。

ひとり1つずつイチゴをとることを1かいとして、とれたかいすうをかぞえます。

F5でRUN、だんだんイチゴがへって行って、ひとり4コと、こたえができました。

イチゴのかずを20コにしてみましょう。

ひとり6コ、イチゴは2コあまるようです。

こうしてあまったかずをひょうじしておくとうわかりやすいです。

ひとり6コずつ、3にん、あまりが2つ。ぜんぶでイチゴはいくつ？

もとのイチゴのかずです。イチゴのかずを200コや2000コにかえてみましょう。

30から50までのくりかえしを、しゅんじにやってくれるのがわりざんです。

「/」はわりきれないぶんはきりすて。「%」はあまりをけいさんします。

やってみよう！

1. イチゴのかずを、にんずうよりすくなくするとどうなるでしょう？
2. わけるにんずうを5にんにかえてみよう（0にんでは？）
3. イチゴ2000コを7にん、くりかえしをつかったプログラムをつくってみよう
4. 50000コのイチゴを5にんでわけたらひとりいくつ？

公倍数さがし（算数） - IchigoJam

1から100までの数を表示しよう

```
10 FOR X=1 TO 100↵
20 ?X↵
30 WAIT 3↵
40 NEXT↵
```

FOR/NEXT（フォー・ネクスト）コマンドで、変数Xを1ずつ増やしながら、100になるまで、くりかえし。

余りを求める記号「%」を使って、3の倍数だけ、表示しよう

```
20 IF X%3=0 ?X↵
```

5の倍数も表示するとき使う「どちらか」を意味する「OR（オア）」

```
20 IF X%3=0 OR X%5=0 ?X↵
```

3の倍数で、5の倍数でもあるものを表示するとき使う「両方」を意味する「AND（アンド）」

```
20 IF X%3=0 AND X%5=0 ?X↵
```

これが3と5の「公倍数」、はじめの数 15 が「最小公倍数」

2と3と5の「公倍数」を表示しよう

```
20 IF X%2=0 AND X%3=0 AND X%5=0
?X↵
```

100までに「公倍数」が3つ、「最小公倍数」は30

やってみよう！

1. 7の倍数を表示しよう
2. 13の倍数を表示しよう
3. 7の倍数と13の倍数、両方表示しよう
4. 7と13の公倍数を表示しよう
5. 1から1000までに3と7の公倍数がいくつあるか、すばやく数えよう

公約数さがし（算数） - IchigoJam

182は何で割り切れるかな？

```
10 N=182↵
20 FOR X=2 TO N-1↵
30 IF N%X=0 ?X↵
40 NEXT↵
```

FOR/NEXT（フォー・ネクスト）コマンドで、変数Xを2から1ずつ増やしなが
ら、くりかえし。N=182 は LET N,182 としてもOK！

でてきた数「約数」をチェックしてみよう

```
?182%91↵
0↵
?182%26↵
0↵
```

ちゃんと割り切れますね！

455は何で割り切れるかな？

```
10 N=455↵
```

182と455、どちらも共通して、割り切れる数を表示

```
10 N=182:M=455↵
20 FOR X=2 TO N-1↵
30 IF N%X=0 AND M%X=0 ?X↵
40 NEXT↵
```

表示された数 7,13,91 が、182と455の「公約数」

一番大きな数 91 が、182と455の「最大公約数」

やってみよう！

1. 3432と2002の「公約数」を表示しよう
2. 3432を「最大公約数」で割ってみよう
3. 2で計算した数と、「最大公約数」をかけてみよう
4. 2002の「約数」がいくつあるか数えよう
5. 4桁の数で「約数」がない数「素数」をさがそう
6. 「素数」同士には「公約数」がないことを確認しよう

四捨五入（算数） - IchigoJam

切り捨てよう！

41は40。44は40。45も40。49だって40。1の位を切り捨てるといいます。

```
?44/10⇐  
4⇐
```

整数の計算になるので、10で割ったら 4.4 ではなく 4 になります。

```
?44/10*10⇐  
40⇐
```

10で割って、10をかけると、なんと切り捨てのできあがり！

切り上げよう！

49は50。45は50。44、41も50。40は40。1の位を切り上げるといいます。

```
?44/10*10+10⇐  
50⇐
```

切り捨てた後、10を足せば切り上げのできあがり！？

```
?40/10*10+10⇐  
50⇐
```

あら、40も50になってしまいました。1の位で判断させてみましょう。

```
10 INPUT N⇐  
20 LET M,N%10⇐  
30 ?M⇐
```

「%」は、IchigoJam BASICの世界では、余りを計算する記号。

もし1の位が0なら切り捨てて、1以上だったら切り上げる。

```
40 IF M=0 ?N/10*10⇐  
50 IF M>=1 ?N/10*10+10⇐
```

やってみよう！

1. 72、1の位をプログラムで切り上げよう。
2. 1の位が4以下で切り捨て、5以上で切り上げる四捨五入プログラムを作ろう
3. $?(N+9)/10*10$ でも1の位の切り上げができることを確認しよう
4. 1の位を四捨五入するプログラムを作ってみよう

分数の足し算（算数） - IchigoJam

分数は、割り算。二分の一は、 $1/2$ と表します。

$A/B+C/D$ を計算するプログラム、まずは分数をINPUTを使ってセットします

```
10 ? "A/B+C/D=" ;  
20 INPUT "A?", A;  
30 INPUT "B?", B;  
40 INPUT "C?", C;  
50 INPUT "D?", D;
```

4つの数を入力して、エンター。?A、などで確認しましょう。

答E/Fを計算してみましょう。分母をそろえて、分子同士を足します。

```
60 F=B*D:E=A*D+B*C;  
70 ?E;" / ";F;
```

計算できました！ $1/2+1/3$ を計算してみましょう。

$1/3+1/3=6/9$ 、計算は合っていますが約分したいですね

```
65 FOR I=F TO 2 STEP -1;  
66 IF F%I=0 AND E%I=0 F=F/I:E=E/  
/I;  
67 NEXT;
```

まず分母の数で分子と分母を割って、割り切れるかチェックして、両方割り切れたら割って通分します。これを数を小さくしながら2まで繰り返します。

$1/2+3/2=2/1$ 、分母が1の時は「一分の」を消しましょう。

```
70 IF F=1 ?E ELSE ?E;" / ";F;
```

やってみよう！

1. いろいろ計算させてみよう
2. 答がおかしくなる数を調べてみよう（IchigoJamの数は-32768から32767）
3. 分数の引き算プログラムを作ってみよう
4. 分数のかけ算、割り算のプログラムを作ってみよう

分数のかけ算（算数） - IchigoJam

分数は、割り算。二分の一は、1/2と表します。

$(A/B)*(C/D)$ を計算するプログラム、まずは分数をINPUTを使ってセットします

```
10 ?" (A/B)*(C/D) =" +  
20 INPUT "A?", A+  
30 INPUT "B?", B+  
40 INPUT "C?", C+  
50 INPUT "D?", D+
```

4つの数を入力して、エンター。?A、などで確認しましょう。

答E/Fを計算してみましょう。 $(A/B)*(C/D)=A/B*C/D=(A*C)/(B*D)$ です。

```
60 E=A*C:F=B*D+  
70 ?E;" / ";F+
```

計算できました！ $(2/3)*(1/4)$ を計算してみましょう。

2/12、計算は合っていますが約分したいですね

```
65 FOR I=F TO 2 STEP -1+  
66 IF F%I=0 AND E%I=0 F=F/I:E=E  
/I+  
67 NEXT+
```

まず分母の数で分子と分母を割って、割り切れるかチェックして、両方割り切れたら割って通分します。これを数を小さくしながら2まで繰り返します。

$(4/3)*(3/2)=2/1$ 、分母が1の時は「一分の」を消しましょう。

```
70 IF F=1 ?E ELSE ?E;" / ";F+
```

やってみよう！

1. いろいろ計算させてみよう
2. 答がおかしくなる数を調べてみよう（IchigoJamの数は-32768から32767）
3. 分数の割り算プログラムを作ってみよう
4. 分数の足し算、引き算のプログラムを作ってみよう

フィボナッチすうれつ - IchigoJam

「かず」が「れつ」のように、つらなったものが「すうれつ」。

```
10 N=0↵
20 N=N+1↵
30 ?N;" ";↵
40 GOTO 20↵
```

20ぎょうにいくたびに、まえのかずに1たします。(LET N,N+1 とおなじ)
いちずつふえる、すうれつのできあがり！

さいしょは0、つぎは1。

そのつぎからのかずは、ひとつまえのかずと、ふたつまえのかずをたしたものの。
という、「フィボナッチすうれつ」をつくってみます。

```
10 A=0:B=1:"0 1 ";↵
20 C=A+B:A=B:B=C↵
30 ?C;" ";↵
40 GOTO 20↵
```

Aがふたつまえ、Bにひとつまえをいれたので、つぎのかずCはA+B。

つぎのすうれつをけいさんするために、AにBを、BにCをいれてくりかえし。
どんどんおおきくなって、おかしくなってしまうところでとめましょう。

```
25 IF C<0 END↵
```

フィボナッチすうれつは、ひまわりのたねや、まきがいなど、しぜんにあわられるふしぎなすうれつ。いろんなえにつかってみるとおもしろいかも！？

0と1でかずをあらわす、2しんすうにしてみましょう

```
30 ?BIN$(C)↵
```

やってみよう！

1. さいしょが0で、3ずつたすすうれつをつくろう
2. さいしょが1で、どんどん2ばいにするすうれつをつくろう
3. 2のすうれつを2しんすうでみてみよう
4. みっつまえまでたしてつくる、トリボナッチすうれつをつくろう
5. オリジナルなすうれつをつくって、なまえをつけよう

はじき (はやさ、じかん、きより) けいさんき - IchigoJam

はやさのたんいは、メートル/びょう。じかんは、びょう。きよりは、メートル
まずは、はやさ・じかん・きよりのどれをけいさんしたいかえらぼう。

```
10 INPUT "HAYASA[m/s]:1 JIKAN[s]  
:2 KYORI[m]:3.",A  
20 IF A<1 OR A>3 GOTO 10
```

はやさ(メートル/びょう)は、きより(メートル) わる(÷) じかん(びょう)。
きよりとじかんをいれて、はやさをけいさん。

```
30 IF A=1 INPUT "KYORI[m]?",K:IN  
PUT "JIKAN[s]?",J:?"HAYASA[m/s]=  
";K/J
```

じかん(びょう)は、きより(メートル) わる(÷) はやさ(メートル/びょう)。
きよりとはやさをいれて、じかんをけいさん。

```
40 IF A=2 INPUT "KYORI[m]?",K:IN  
PUT "HAYASA[m/s]?",H:?"JIKAN[s]=  
";K/H
```

きより(メートル)は、はやさ(メートル/びょう) かける(×) じかん(びょう)。
はやさとじかんをいれて、きよりをけいさん。

```
50 IF A=3 INPUT "HAYASA[m/s]?",H  
:INPUT "JIKAN[s]?",J:?"KYORI[m]=  
";H*J
```

やってみよう！

1. 200メートルのきよりを、60びょうですすんだときのはやさは？
2. 3メートル/びょうのはやさで、60メートルのきよりをすすんだときのじかんは？
3. 10メートル/びょうのはやさで、30びょうすすんだときのきよりは？
4. ランダムできよりとじかんをきめて、はやさをあてるゲームにしよう

角度と多角形

座標(0,0)から(63,47)へ、線を引いてみよう

```
DRAW 0,0,63,47
```

画面の真ん中から、上に向かって10だけ歩こう

```
10 CLS : X=32 : Y=24 : L=10
20 A=X
30 B=Y-L
40 DRAW X,Y,A,B
```

角度T、まっすぐ右を0度、時計回りに大きくすると上は-90度

```
15 T=-90
```

角度から横方向の距離を教えてくれるCOS

角度から縦方向の距離を教えてくれるSIN

```
20 A=X+COS(T)*L/256
30 B=Y+SIN(T)*L/256
```

一歩ごとに右に90度曲がってみよう

```
16 D=90
50 X=A : Y=B : T=T+D
60 WAIT 30 : GOTO 20
```

4回曲がると元の場所

やってみよう！

1. 16行のD、-90にするとどうなるかためそう
2. Dをいろんな数に変えてみよう
3. 星を描いてみよう
4. 15行のTを変えると何が変わるかためそう
5. 好きな図形を好きな場所に描いてみよう

AからZまで、はやうちゲーム

もじをおおきくしてみよう

VIDEO3

AからZまで、なんびょうでうてるか、タイムアタック！

```
10 N=65:CLT
20 ?CHR$(N);
30 IF INKEY() != N GOTO 30
40 N=N+1:IF N!=91 GOTO 20
50 ? : ?TICK() / 60
```

うちこみおわったら、F5でRUN

AからZまで、26コ。うったらタイムをびょうでひょうじするよ
とちゅうでやめるときは、ESC、もういちどちょうせんは、F5

きわどいあらそい、じかんをこまかくひょうじしたいなら

```
50 T=TICK()
60 ? : ?T / 60 ; " . " ; T / 6 % 10
```

こもじをひょうじするかいぞうほう

```
20 ?CHR$(N+32);
```

もじはコンピューターのなかでは、かずでかんりされてるよ

アルファベットのAは、65

```
?CHR$(65)
A
?ASC("A")
65
```

やってみよう！

1. A、B、Cと、こえをだしながらやってみよう
2. じぶんのきろくをメモしよう
3. なんかいめで、30びょうをきれるかな？
4. ともだちとヨーイドンでたいせんしてみよう
5. りょうてをつかったブラインドタッチができるとかっこいい

えいごのかずをマスターしよう

えいごで1、2、3は、ワン、ツー、スリー！

```
10 LET[1],"ONE","TWO","THREE"↵
20 N=RND(3)+1↵
30 ?STR$(N)↵
```

F5をおすたびに、1か2か3がでるよ

ひょうじされたら、すばやくすうじキーでこたえるゲームにしよう！

```
15 CLS:WAIT60:LC 0,10:CLT↵
40 IF INKEY()-48<>N GOTO40↵
50 ?TICK()↵
```

こたえるまでのじかんがでるよ。60、きれるかな？

なれたら、4から9もくわえよう

```
11 LET[4],"FOUR","FIVE","SIX","SEVEN",  
"EIGHT","NINE"↵
20 N=RND(9)+1↵
```

もんだいを10かいだして、ごうけいタイムをだしてみよう。

```
14 S=0:FOR I=1 TO 10↵
50 T=TICK():?T:WAIT30:S=S+T↵
60 NEXT↵
70 ?"TIME:";S↵
```

500をきれるかな？

やってみよう！

1. タイム、400きれるかな？
2. もんだいすうを5もんにしてみよう
3. がめんのまんなかにもんだいをひょうじしてみよう
4. スコアひょうじのいちをかえてみよう
5. もんだいにONEしかでないようにしてみよう

えいごのあいさつをマスターしよう IF

INPUT（インプット）コマンドをつかって、じかん（TIME）をにゅうりよく！

```
10 INPUT A
20 PRINT "TIME:";A
```

ごぜん・ごごをえいごでおしらせしてくれるようにしよう。

```
30 IF A<12 PRINT"AM"
40 IF A>=13 ?"PM"
```

ごぜんはAM、ごごはPM。にほんごは、0~24じ（ごぜん0~12じ、ごご13~24じ）
だけど、えいごでは、AM0~12じ、PM0~12じとひょうきするんだ。

もし、21じ=PM9じになったらこうひょうじしよう。

```
50 IF A=21 ?"MOU9JI!"
```

おきていると、おとなにおこられるよ！もうねよう。

かずを、にほんごのあいさつにあてはめよう！4つにわかれるよ。

```
60 INPUT "0:OHAYOU 1:KONNICHIHA 2:KONBANHA 3:OYA
SUMI ",S
```

もし、AM0~5じにゲームをしていたとしたら、どうなるかな？

```
70 IF A>=0 AND A<=5 ?"HAYAKUNENASAI!"
```

おこられた！ねます！ でもあとちょっと！えいごであいさつできるかな？

```
80 IF A>=6 AND A<=11 AND S=0 ?"GOOD MORNING!"
90 IF A>=12 AND A<=17 AND S=1 ?"GOOD AFTERNOON
🌞"
100 IF A>=18 AND A<=20 AND S=2 ?"GOOD EVENING:)"
110 IF A>=21 AND A<=23 AND S=3 ?"GOOD NIGHT!🚀"
```

やってみよう！

1. じかんごとにちがうメッセージをせっていしてみよう
2. TODOメモとして、やることをひょうじさせてみよう
3. にほんごのあいさつをカタカナひょうきにしてみよう

たんどてすと

えいごのたんどもんだいをつくってみよう！

「かな」キー（またはCTRL+SHIFT）をおして、AIUEOとうつとカタカナができるのは知っているかな？

```
アイウエオ
```

ねこはえいごで"cat"だね！

がめんにひょうじしてみよう

```
? "ネコ ハ エイゴ テ cat テス"
```

ねこのえいごで、くいずをつくってみよう！

```
10 ? "ネコ ハ エイゴ テ ナニ テスカ?"  
20 ? "1:DOG 2:CAT 3:KAT"
```

3たくもんだいにしてみよう！

くいずのこたえを入力してもらおう！

せいかいだったらSEIKAI!

はずれだったらHAZURE!

と表示しよう！

```
30 INPUT A  
40 IF A=2 ? "SEIKAI!" ELSE ? "HAZURE!"
```

ほかのたんどだとどうなるかな？

```
10 ? "リンゴ ハ エイゴ テ ナニ テスカ?"  
20 ? "1:APPLE 2:APPRE 3:ORANGE"  
30 INPUT A  
40 IF A=1 ? "SEIKAI!" ELSE ? "HAZURE!"
```

やってみよう！

1. じぶんでたんどをかんがえてやってみよう！
2. ほかのせんとくしをどうするかかんがえてみよう
3. せんとくしをふやしてみよう
4. むずかしいたんどにもちようせんしてみよう

ローマじでうつ、カタカナ

「かな」キーをおして、AIUEOとうつとカタカナがでる（またはCTRL+SHIFT）

アイウエオ

「CAPS」キーか「英数」キーをおすと、もとにもどる（またはCTRL+SHIFT）

「かな」キーをおして、KA KI KU KE KO とうってカキクケコ（カぎょうはK）

カ キ ク ケ コ

さぎょうはS、AIUEOとくみあわせよう、SA TI NU HE MO YA RI WA

サ チ ツ ネ モ ヤ リ ワ

だくてん、はんだくてんは、GZDBPをつかうよ、GA ZI DU BE PO

カ^ゞ シ^ゞ ツ^ゞ ヘ^ゞ ホ[□]

「を」はWO、「ん」はN だけど、すぐにだしたいときはNN

ヲ ン

ちいさい「っ」は、まえのじをふたつかく YATTA （LTUでもだせる）

ヤ ッ ヲ

Yをつかって、ちいさい「やゆよ」KYA KYU KYO （LYAでもだせる）

キャ キュ キョ

のばすもじは「-（ひく）」、まるは「.（ドット）」、てんは「,（コンマ）」

ー 〃 丶

ちいさい「アイウエオ」は、LA LI LU LE LO でだせる

アイウエオ

やってみよう！

1. Sをつかって「サシスセソ」とかいてみよう
2. 「タチツテト」とかいてみよう
3. KYOUHA,TANOSIKATTADESU. とかいてみよう
4. じぶんのなまえをかいてみよう
5. すきなたべものをかいてみよう

ローマじ、アからンまではやうちゲーム

もじをおおきくしてみよう

VIDEO3

アからンまで、なんびょうでうてるか、タイムアタック！

```
10 N=177:CLT
20 ?CHR$(N);
30 IF INKEY() != N GOTO 30
40 N=N+1:IF N != 222 GOTO 20
50 ? : ?TICK() / 60
```

うちこみおわったら、F5でRUN

アからンまで、46コ。うったらタイムをびょうでひょうじするよ
とちゅうでやめるときは、ESC、もういちどちょうせんは、F5

きわどいあらそい、じかんをこまかくひょうじしたいなら

```
50 T=TICK()
60 ? : ?T / 60 ; " . " ; T / 6 % 10
```

すきなすうじをもじにかえてみよう！

```
?CHR$(100)
d
```

もじはコンピューターのなかでは、かずでかんりされてるよ 100はもじになおすと"d"だね！ ほかのすうじでもたくさんためしてみよう！

はんたいにすきなもじをすうじにしてみよう！

```
?ASC(" G ")
71
```

Gはすうじになおすと71だね！ ほかにもたくさんためしてみよう！ なにかすうじともじのあいだにかんけいせいがあるかも！？

やってみよう！

1. ローマ字をべんきょうしながらやってみよう
2. じぶんのきろくをメモしよう
3. ともだちとヨーイドンでたいせんしてみよう

4. コンピューターのなかのすうじともじのかんけいをしらべよう

5. りょうてをつかったブラインドタッチができるとかっこいい

CC BY IchigoJamプリント <http://ichigojam.net/print/> IchigoJam@jig.jp

はじめてのかいろ OUT (アウト)

でんきのとおりみちのことを「かいろ」といいます。LEDを2つよういして、かいろをつくってひからせてみましょう。

IchigoJamのほんたいのあるピンソケットCN4、14コのあなのそれぞれのやくめがシールにかいてあります。LEDのながいほうのあしをOUT1へ、みじかいほうのあしをGNDへ、それぞれさしこみましょう。

OUT1,1

「OK (オーケー)」とでて、さしこんだLEDがひかったら、だいせいこう！ OUT1のピンの「でんあつ」がたかくなって、LEDのなかを「でんりゅう」がとおり、GND (グラウンド、でんあつ0) へながれることでひかります。

OUT1,0 (アウト、ゼロ)、エンターでけせます。

OUT1,0

もうひとつLEDをさしこんでみましょう。LEDのながいほうのあしをOUT2へ、みじかいほうのあしをOUT3へさしこみます。

OUT2,1

OUTコマンド、さいしょのかずがピンのばしょ、つぎのかずでつけるか、けすかをきめます。

こうごに、てんめつさせてみましょう。

```
10 OUT1,1:OUT2,0:WAIT30+
20 OUT1,0:OUT2,1:WAIT30+
30 GOTO10+
RUN+
```

とめるときは [esc] (エスケープ キー)

やってみよう！

1. OUT2,1 でLEDをつけたあと、OUT3,1 とやってみよう
2. そのあと OUT3,0 でLEDがつくわけをかんがえてみよう
3. IchigoJamほんたいについているLEDとあわせててんめつさせよう
4. 3つのLEDがじゅんばんにぜんぶつくプログラムをつくろう
5. LEDをぜんぶつけてから、OUT0 または F7 をおしてみよう
6. OUT3,1でLEDがひかるようにかいろをつくりかえてみよう

ほこうしゃしんごうをつくる - IchigoJam

みどりかあおのLEDを1つ、IchigoJamのピンソケットCN4に、LEDのながいほうのあしをOUT4へ、みじかいほうのあしをOUT3へさしこみます。

```
OUT4,1↵
OK↵
OUT4,0↵
OK↵
```

ひかりましたか？

ひからなかったらさしこむばしょをよくかくにんしましょう

しばらくひからせてけしてみましよう

```
OUT4,1:WAIT480:OUT4,0↵
```

さいしょはあおしんごう、しばらくたったら、あかしんごう、くりかえし！

```
10 LED0:OUT4,1↵
20 WAIT480↵
30 LED1:OUT4,0↵
40 WAIT480↵
50 GOTO10↵
RUN↵
```

おんがくをならしてみましよう

```
15 PLAY"O4L8 E2E4D4E4ED>B2<F4FFA
4FEFEDDE2"↵
```

* O4でオクターブ、L8でひょうじゅんのおとのながさせってい

やってみよう！

1. あかしんごうのじかんをながくしてみよう
2. あおしんごうのてんめつをつくろう
3. おんがく「とおりゃんせ」のつづきをつくろう
4. おんがくのながさにあわせて、あおしんごうのじかんをちょうせいしよう
6. BTN()をつかっておしぼたんしんごうにしてみよう
7. あおしんごうのながさをせっていできるようにしよう

サーボをあやつる PWM (ピーダブリューエム)

ロボットのうでなどでつかう、サーボをあやつろう。サーボと、ジャンパーせん3ほん、ピンソケットCN5がついたIchigoJamをじゅんびしよう。

・サーボをつかったかいり

サーボのあか：IchigoJamのCN5の5V（いちばんうえ）へ

サーボのちゃ：IchigoJamのCN5のGND（いちばんした）へ

サーボのオレンジ：IchigoJamのOUT2へ

```
PWM2,100+
```

「OK（オーケー）」とでて、サーボがギツとうごいたら、だいせいこう！
さいしょのかずでOUT2（2〜5、OK）、つぎのかずでかくどをしてい。

かくどは50から200のかずでしていできる。

```
PWM2,50:WAIT60:PWM2,200+
```

WAITコマンドとあわせてつかって、いろいろじどうでうごかそう！

ボタンをおしたら、3かいでをふるロボットをつくります。

```
10 IF BTN(<)=0 GOTO 10+
20 FOR I=1 TO 3+
30   PWM2,50:WAIT30+
40   PWM2,200:WAIT30+
50 NEXT+
60 GOTO 10+
```

IF（イフ、もし）BTNがおされていなければ、GOTO（ゴートゥー）10でくりかえし、ボタンがおされたらつぎへ。FOR（フォー）はかいすうつきくりかえし、NEXT（ネクスト）までを3かいくりかえします。

やってみよう！

1. てをゆっくりふるようにかいぞうしよう
2. てをちいさくふるようにかいぞうしよう
3. てはどれだけはやくふれるかじっけんしよう
4. サーボのちゃをOUT3にさしてうごかしてみよう
5. ボタンをおして3びょうごに、1どてをふるプログラムをつくろう

0か1かのにゅうりよく IN (イン)

ジャンパーせんを2ほんよういしよう。

IchigoJamのピンソケットCN3とCN4の14コのあな、1ほんをIN1へ、もう1ほんをGNDへさしこみます。

```
10 ?IN(1):WAIT10:GOTO10↵  
RUN↵
```

ずっとひょうじされる、1。ジャンパーせんの2ほんのきんぞくぶぶんをくつけると0にかかります。

0か1かのにゅうりよくをそのままLEDコマンドにわたしてみましよう。

```
10 LED IN(1):GOTO10↵
```

0を1へ、1を0にはんてんする!きごうをつけると、ぎゃくになる。

```
10 LED !IN(1):GOTO10↵
```

0になるまでまってもらおう

```
10 IF IN(1)=1 GOTO 10↵
```

はなしているとき、IN(1)が1なので、ずっと10ぎょうをぐるぐるくりかえして、0になったらおわって、OKがでる。

0になるまでまって、1になるまでまってをくりかえす

```
10 IF IN(1)=1 GOTO 10↵  
20 ?"0!"↵  
30 IF IN(1)=0 GOTO 30↵  
40 ?"1!"↵  
50 GOTO 10↵
```

やってみよう!

1. ジャンパーせんがふれたら、おとがなるようにしてみよう
2. ジャンパーせんをふれたじょうたいにしておき、はなれたらけいこくおんをならそう
3. IN(1)が0のときだけLEDをてんめつさせてみよう
4. ジャンパーせんをIN4にさしかえ、IN(4)をつかってじっけんしてみよう
5. IN2とIN3はそのままではうまくうごかないことをたしかめよう (プルアップていこうがひつよう)

りょうのにゅうりょく ANA (アナ)

ひかりセンサー、おんどセンサー、ボリュームなどをIchigoJamのIN2にせつぞく。

0から1023までのかずでひょうじされます。

```
10 ?ANA(2):WAIT10:GOTO10↵  
RUN↵
```

センサーをいろいろいじって、かずがどうかわるかかんさつしてみよう。

かずが500よりおおきいときにLEDをひからせるには

```
10 ?ANA(2):LED ANA(2)>500:GOTO10↵  
↵
```

かずが400よりちいさいときにLEDをひからせるには

```
10 ?ANA(2):LED ANA(2)<400:GOTO10↵  
↵
```

かずが800よりおおきくなるまでまつには

```
10 ?ANA(2):IF ANA(2)<800 GOTO10↵
```

ANA(2)が800よりちいさいときは、ずっと10ぎょうをぐるぐるくりかえして、こえたらおわって、OKがでる。

500よりおおきくなるまでまってBEEP、400よりちいさくなるまってBEEPをくりかえす

```
10 ?ANA(2):IF ANA(2)<500 GOTO10↵  
20 BEEP 5↵  
30 ?ANA(2):IF ANA(2)>400 GOTO30↵  
40 BEEP 15↵  
50 GOTO 10↵
```

やってみよう！

1. ひかりセンサーをつかって、くらくなったらLEDをつけよう
2. おんどセンサーをつかって、あたたかくなったらLEDをつけよう
3. IchigoJamのBTNにセンサーをつないで、ANA(0)であたいがかわるのをみてみよう
4. 2つのセンサーがりょうほうたかいときにおとをならそう
5. センサーのあたいでグラフをかいてみよう

じどううんてんをつくるOUTとANA

ロボットをうごかそう（カムロボ、らでいっしゅ、paprikaなど）

```
1  OUT33
```

1. 「F3、0、エンター」で、0にSAVE
2. IchigoDakeをぬく
3. ロボットのスイッチオン
4. ロボットにIchigoDakeをボタンをおしながらさして、てをはなす
5. とめるときは、ロボットのスイッチをオフ、IchigoDakeをぬく

3びょうすすんでとまる

```
2  WAIT180:OUT0
```

6. 「F2、0、エンター」で、0をLOAD
7. 「F4」でプログラムをかくにん
8. プログラムをついか
9. GOTO 1（うえてやった、てじゅん1へ）

「3びょうすすんで1びょうとまる」をGOTOつかって、くりかえし

```
3  WAIT60:GOTO1
```

じどうブレーキをつくろう

センサー(ANA(2))がはんのうするまでまって、OUT0でとめる

```
2  IF ANA(2)<600 GOTO2  
3  OUT0
```

やってみよう！

1. ぜんごにうごかそう（まえ:OUT33、バック:OUT18）
2. しかくをえがくようにうごかそう（みぎ:OUT17、ひだり:OUT34）
3. じぐざぐにはしらせよう
4. センサーにはんのうしたらバックしよう
5. センサーにはんのうしたらまがってよけよう
6. かべをみぎやひだりによけながらすすみつづけるロボットをつくろう

サーボとセンサーで電子工作

ボタンを押したらサーボの動く速さが変わる（サーボ SG90など）

```
10 PWM 2,80:WAIT 10+BTN()*30
20 PWM 2,140:WAIT 10+BTN()*30
30 GOTO10
```

（PWMへの値は70～220にする、範囲を超えると壊れちゃうかも）

ボタンを押したら動き出す

```
1 IF BTN()=0 CONT
```

（CONT は GOTO LINE() と同じ、その行を繰り返し）

センサー(IN2にさした距離センサーなど)の値によって、速さが変わる

```
10 PWM 2,80:WAIT 10+ANA(2)/10
20 PWM 2,140:WAIT 10+ANA(2)/10
25 ?ANA(2)/10
30 GOTO10
```

変数(A)を使ってすっきり

```
1 IF BTN()=0 CONT
20 A=ANA(2)/10
30 PWM 2,80:WAIT A
40 PWM 2,140:WAIT A
50 GOTO 20
```

センサーの値が400より小さくなったら、ボタン待ちへ


```
25 IF A<400 GOTO1
```

やってみよう！

1. ボタンを押したら、3秒止まるロボをつくろう
2. くりかえしのコマンドFORを使って、5回動くロボをつくろう
3. LCとPRINTを使って画面の動きと連動させよう
4. VIDEO2/VIDEO3/VIDEO5などを使って画面を派手にしよう（F8で戻る）
5. BEEPやPLAYを使って音と連動させよう

2だいのネットをつくろう

IchigoJamを2だい、ジャンパーせんを3ほん、よういして、つなぎます。



※GNDとVCCをまちがうとショートしてこわれたりするので、きをつけて！

IchigoJam1でつぎのようにコマンドをうちこんでみましょう。

```
? "LED1+
```

IchigoJam2になにがおこりましたか？

なにもおきないときはせんがちゃんとつながっているか、かくにんしよう。

```
? "LED0+
```

メッセージをおくってみます

```
? "HELLO+
```

HELLOというコマンドはないので、Syntax error（コマンドまちがい）となり、そのエラーがおくられてまた Syntax error とくりかえします。TXDまたはRXDにさしたジャンパーせんを、いちどぬいてとめましょう。

コマンドをむこうかするコマンド「」（アポストロフィ、Shift+7）/REM」

```
? " /HELLO+
```

ローマじでカタカナをつかうには「かな」キーできりかえます。

```
? " /ハロー+
```

もういちど「かな」キーをおすと、アルファベットモードにもどります。

やってみよう！

1. ふたりでおしゃべりしてみよう
2. いちどせんをぬいて、もういちどつないでみよう
3. TXD-RXDのせんを1ほんぬくとどうなるかじっけんしよう
4. ともだちのパソコンをえんかくそうさしよう
5. つうしんたいせんゲームをかながえよう

Ichigoチョコでめいろだっしゅつ - IchigoJam

くみたてたIchigoチョコを、キーボードとビデオケーブルをつなぐ。
IchigoJamのスイッチをON、IchigoチョコのスイッチをPROGにする。
OUT (アウト) コマンドでまえにすすむよ！

```
OUT6➡
```

とめたいときは、F7キー(OUT0)

WAITコマンドをつかって、1びょうすすんで、じどうきにとめよう

```
OUT6:WAIT60:OUT0➡
```

WAITのかずをかえると、すすむながさを、ばいにしてみよう

まっすぐなみちをはして、ゴールぴったりでとめるプログラムをつくろう

```
10 OUT6:WAIT180➡
20 OUT0➡
```

F5キー(RUN)でうごきをかくにんしたら、F3キー(SAVE)、0でほぞん！
IchigoチョコのスイッチをOFFにして、キーボードとビデオケーブルをぬく。
スタートにIchigoチョコをおいたら、IchigoチョコのスイッチをRUN！
(IchigoチョコのRESETボタンでリトライ)

IchigoチョコのスイッチをOFF、ケーブルをつないで、スイッチをPROGへ
F2キー(LOAD)、0、でよみこんで、プログラムをついかしよう

```
15 OUT4:OUT60➡
```

F4キー(LIST)でプログラムがかいぞうできたか、かくにん、F5キー(RUN)！

```
15➡
```

プログラムのばんごうだけかいてエンターで、そのぎょうをけせるよ！

やってみよう！

1. みぎかいてん OUT2 で、5びょうかんまわるロボットをつくろう
2. ゴールにいったらUターンするロボットをつくろう
3. ランダムコマンドをつかって、OUT RND(7) とするとどうなる？
4. OUT2,1 や OUT2,0 や OUT3,1 や OUT3,0 でおきることをかいいいしよう
5. じぶんでめいろをつくって、かれいにだっしゅつしよう！

おはなLEDでルーレットゲーム

おはなLEDシールド「Flower LED Shiled for IchigoJam」をはんだづけして、くみたてたら、IchigoJamにドッキング！（おはなLEDがうえ、ボタンがしたになるように）

おはなのちゅうおうがひかったらせいこう！

LED1✚

IN1、2、4をアウトプットにきりかえて、OUT8、9、11としてつかおう

OUT8,1:OUT9,1:OUT11,1✚

（もし、もとのインプットにもどしたいときは OUT8,-1 とかでOK）

ぐるぐるまわって、ボタンでとめるルーレット

```
10 LET N,1:OUT0✚
20 LET F,0:LET T,2✚
30 OUT N,1:WAIT T✚
40 IF BTN() F=1✚
50 IF F LET T,T+T/2✚
60 IF T>60 GOTO 120✚
70 OUT N,0✚
80 LET N,N+1✚
90 IF N=12 LET N,1✚
100 IF N=7 OR N=10 LET N,N+1✚
110 GOTO 30✚
120 LED 1✚
130 IF BTN()=0 GOTO 130✚
140 IF BTN() GOTO 140✚
150 GOTO 20✚
```

RUNするとぐるぐるまわるよ！

やってみよう！

1. とまるまでのながさをながくしてみよう
2. ゆっくりまわるようにしてみよう
3. いちばんしたにとまったときに「アタリ」とひょうじしよう
4. まわっているときにも、いちばんしたのときLEDをつけてみよう
5. ぎゃくかいてんさせてみよう

PL9823でいろあそび

PL9823はコンピューターつきのフルカラーLED。いちばんながいピンがみぎから2ばんめにくるようにみて、ひだりからじゅんに、IchigoJamのOUT1、VCC、GNDにつないだら、つぎのマシンごをうちこもう。

```
10 POKE #700,240,181,114,182,64,  
24,80,33,9,2,1,49,9,4,4,136,1,48  
,128,37,1,48,3,120,255,34,74,96,  
3,39,10,38,43,66,1,208,10,39,3,3  
84  
20 POKE #728,1,63,253,209,0,34,7  
4,96,1,62,253,209,109,8,239,209,  
1,60,234,209,98,182,240,1894
```

まちがえていると、とまってしまうので、SAVEしてからRUNがオススメ！

1つ、あかくひからせてみよう！

```
POKE#800,3,0,50,0,0:?USR(#700,#8  
00)4
```

みどり、あお

```
POKE#802,0,50,0:?USR(#700,#800)4  
POKE#802,0,0,50:?USR(#700,#800)4
```

もうひとつPL9823をよういして、1つめのいちばんみぎのピンから、2つめのいちばんひだりのピンへつなぐ。ひだりから2ばんめ3ばんめはVCC、GNDをつないで、2つまとめてみどりいろにひからせよう

```
R=0:G=50:B=0:POKE#800,3*2,0,R,G,  
B,R,G,B:U=USR(#700,#800)4
```

おなじようにじゅずつなぎして、3つひからせるときは3*2を3*3にして、R,G,Bを3セットかこう！

やってみよう！

1. 2つともあかくひからせよう
2. 2つをほこうしゃしんごういろにひからせよう
3. むらさき、しろいろにひからせてみよう
4. きいろてんめつしんごうをつくってみよう
5. グラデーションさせてみよう

WS2812Bでいろあそび

WS2812BはコンピューターつきのフルカラーLED。

5VをIchigoJamのVCC、DIをIchigoJamのOUT1、GNDをIchigoJamのGNDにつないだら、つぎのマシンごをうちこもう

```
10 POKE#700,102,32,240,181,114,182,8,36,36,2,9,25,11,76,128,37,11,120,255,34,34,96,3,39,10,38,43,66,1,208,10,39,3,38,1,63,253,209,0,34#
20 POKE#728,34,96,1,62,253,209,109,8,239,209,2,49,1,56,234,209,98,182,240,189,4,0,1,80#
60 U=USR(#700,0)#
```

まちがえていると、とまってしまうので、SAVEしておこう

1つ、あかくひからせてみよう！つぎのぎょうをうちこんだあと、F5でRUN！

```
[1]=10#
```

みどり

```
[0]=10:[1]=0:[2]=0#
```

あお（LETコマンドで3つまとめてセットできるよ！）

```
LET[0],0,0,10#
```

3つまとめてひからせよう

```
R=4:G=10:B=2:LET[0],G,R,B,G,R,B,G,R,B#
```

さいだい34コまでひからせることができるよ！

やってみよう！

1. 3つともあかくひからせよう
2. 3つをしんごういろにひからせよう
3. むらさき、オレンジ、しろいろにひからせてみよう
4. きいろてんめつしんごうをつくってみよう
5. グラデーションさせてみよう

きらきらしよう (WS2812Bのつづき1)

32コつないだWS2812B、ぜんぶ、ひからせる！

```
30 V=10↵
40 LET[0],V,V,V↵
50 FOR I=1 TO 31:LET[I*3],[0],[1]
  ,[2]:NEXT↵
```

Vはひかりのつよさ。ちいさくしたり、ちょっとおおきくしたりしてみよう。

いろいろないろで、ひからせる

```
40 LET[0],RND(V),RND(V),RND(V)↵
70 WAIT10:GOTO40↵
```

あかくひからせる (40ぎょうは「みどり、あか、あお」のじゅん)

```
40 LET[0],0,RND(V),0↵
```

ぜんぶちがういろで、ひからせる

```
50 FOR I=0 TO 31:LET[I*3],RND(V)
  ,RND(V),RND(V):NEXT↵
```

あおグラデーション

```
50 FOR I=0 TO 31:LET[I*3],0,0,(I
  +TICK()/30)*V:NEXT↵
```

しろいろがふってくる

```
50 FOR I=0 TO 90 STEP 3:LET[I],[
  I+3],[I+4],[I+5]:NEXT:N=RND(V):L
  ET[93],N,N,N↵
```

やってみよう！

1. みどりグラデーションをつくってみよう
2. こうそく、きいろグラデーションをつくってみよう
3. いろんないろをふらせよう
4. キラキラさせてみよう
5. オリジナルなでんしよくをつくってみよう

にじをつくろう (WS2812Bのつづき2)

ひかりのさんげんしょく、あか・みどり・あお。ぜんぶまぜると？

```
V=20:R=V:G=V:B=V:LET[0],G,R,B:U=USR(#700,0)←
```

Vはひかりのつよさ。ちいさくしたり、ちょっとおおきくしたりしてみよう。

いろのグラデーションきれいな「にじ」は「しきそう」でつくれる！

```
500 D=H/60%6:E=H%60*V/60←
510 IF D=0 R=V:G=E:B=0←
520 IF D=1 R=V-E:G=V:B=0←
530 IF D=2 R=0:G=V:B=E←
540 IF D=3 R=0:G=V-E:B=V←
550 IF D=4 R=E:G=0:B=V←
560 IF D=5 R=V:G=0:B=V-E←
570 RTN←
←
H=40:V=6:GSB500:LET[0],G,R,B:U=USR(#700,0)←
```

Hはいろをかくどであらわす「しきそう」0から360、いろいろかえてみよう。

にじいろにどんどんかわるプログラム

```
V=20:FOR H=0 TO 990 STEP 10:GSB500:LET[0],G,R,B:?H:U=USR(#700,0):WAIT10:NEXT←
```

7つのフルカラーLEDで、にじをつくろう

```
V=20:FOR I=0 TO 6:H=I*50:GSB500:LET[I*3],G,R,B:NEXT:U=USR(#700,0)←
```

やってみよう！

1. にじのはじまりのいろをかえてみよう
2. アニメーションするにじをつくってみよう
3. にじがだんだんてくるプログラムをつくってみよう
4. にじのいろのじゅんばんをぎやくにしてみよう
5. オリジナルなでんしょくをつくってみよう

カラフルドットえをかこう WS2812B 16x16

16x16、256コフルカラーLED(WS2812B)でえをかこう！

WS2812BとIchigoJam、5VとVCC、DINとOUT1、GNDとGNDをつなぎます。

```
10 POKE#700,8,34,18,2,10,68,244,
181,114,182,0,32,9,36,36,2,9,25,
1,36,164,70,30,76,11,120,48,59,6
3,34,19,64,91,0,2,180,1,153,201,
24,11,120,27,9,0,240,29,248,11,1
36,27,10,0,240,25,248,11,120,15,
344
20 POKE#73C,19,64,0,240,20,248,2
,188,97,68,1,48,15,34,16,66,228,
209,2,9,16,42,8,208,31,49,1,35,9
3,66,172,70,26,66,219,209,2,49,1
56,70,216,231,98,182,242,189,128
,37,255,34,34,96,3,39,9,38,43,66
,1,2084
30 POKE#778,9,39,3,38,1,63,253,2
09,0,34,34,96,1,62,253,209,109,8
,239,209,112,71,0,0,4,0,1,804
40 U=USR(#700,0)4
```

まちがっていると、とまってしまうので、SAVEをしておきましょう。

がめんのひだりうえ、16x16のりょういぎのすうじがえになります。Insertキーで、てんめつするカーソルがかわり、うわがきモードが、かきやすい！

```
[0]=#F00:RUN4
```

はいれつの0～63、0～9とA～Fの16しゅるいのもじをつかう、16しんすうで、あか(R)、みどり(G)、あお(B)のじゅんに、つよさをしてい。

```
[1]=#00F:RUN4
```

やってみよう！


1. なまえをかいてみよう
2. しんごうきのえをかこう
3. グラデーションをつくってみよう
4. アニメーションさせてみよう

LPC810をあやつろう

コンピューターLPC810にプログラムをほぞんするプログラム(SAVE1)

```
1  /LPC810 Writer*
110  UART3,2:OUT2,0:OUT1,0:WAIT9:
OUT1,1*
120  CLS:"?":GSB@R*
130  ?"Synchronized":GSB@R*
140  ?12000?:GSB@R:GSB@R*
150  ?"A 0":GSB@R:GSB@R*
160  ?"U 23130":GSB@R*
170  ?"W 268436096 ";L:GSB@R:UART
1,2:LC0,-1:FORI=0TOL-1:?CHR$(PEE
K(A+I));:NEXT:UART3,2:LC0,6:?*
180  ?"P 0 0":GSB@R*
190  ?"E 0 0":GSB@R*
200  ?"P 0 0":GSB@R*
210  ?"C 0 268436096 256":GSB@R*
220  OUT2,1:OUT1,0:OUT1,1:END*
300  @R:K=INKEY():IF K=13 K=INKEY
():WAIT9:RTN ELSE CONT*
```

LPC810にLEDとIchigoJamをつ
ないで、LEDをひからせよう！
VCCとGNDに3Vのでんちをつ
なぐとひかるよ！



```
NEW*
10  POKE#700,0,4,0,16,33,0,0,0,0,
0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,
0,0,0,223,251,255,239,6,72,223,3
3,1,96,6,72*
20  POKE#728,6,73,1,96,6,72,12,33
,1,96,6,72,1,33,129,112,254,231,
0,0,128,128,4,64,192,193,0,64,19
1,255,255,255,0,32,0,160,0,0,0,1
60*
90  A=#700:L=80:LRUN1*
```


sakura.ioでネットにつなごう

sakura.ioのブレイクアウトボードをじゅん
びして(<http://fukuno.jig.jp/1794>)、
IchigoJamとつなぎます。
sakura.ioのLEDがてんめつしはじめ、1ぷ
んほどまつと、1かいてんめつになってせ
つぞくかんりょう！

sakura.io		IchigoJam
きいろ 5V	——	5V
あお SDA	——	SDA (or IN3)
みどり SCL	——	SCL (or EX1)
あか 3.3V	——	VCC (+ / 3.3V)
くろ GND	——	GND (- / 0V)

ネットへデータをそうしんしょう

```
800 LET[0],2593,19457,N,0,0,0,N>  
>8^N^102:?I2CR<79,#800,13,#820,3  
>:RTN#  
N=15:GSB800#
```

0とひょうじされればせつぞくOK。PCやスマホで15がとどいていることをかく
にんしょう。Nにおくりたいかずをいれて、GSB800！

ネットからデータをじゅしんしょう

ネットからせつぞくしたsakura.ioへかずをおくっておきます。

```
810 LET[0],48,48:?I2CR<79,#800,3  
,#820,22):F=[16]!=5:N=[18]:RTN#  
GSB810:?N#
```

おくったかず、うけとれましたか？

あたらしいデータをうけとれたときに、へんすうFが1になることをつかって、デ
ータをうけとったらBEEPでおとをならす、ネットにつながるがっきのできあが
り！

```
10 GSB810:IF F=0 CONT#  
20 BEEP N#  
30 GOTO 10#
```

やってみよう！

1. ボタン(BTN)がおされたら、ネットでつたえよう
2. ネットからさいせいするおんがく(PLAY)をえらんでみよう
3. ゲームのスコアをネットにそうしんしょう
4. ドアがあいたらネットにそうしんするモノをつくってみよう
5. スマホからコントロールするロボットをつくってみよう

かわくだりゲーム - IchigoJam

しょうがいぶつをよけて、かわをくだるゲーム。まずは、じぶんキャラ。

```
10 CLS:X=15↵
20 LC X,5:"0"↵
RUN↵
```

※" は[shift]をおしながら「2 ふ」、? は[shift]をおしながら「/ め」

てきキャラ、とうじょう。

```
30 LC RND(32),23:"* "↵
```

※() は[shift]をおしながら「8 ゆ」と「9 よ」

F5キー（=RUNとおなじいみ）をレンドしてみよう！

プログラムでくりかえし

```
40 GOTO 20↵
```

はやすぎた？ [esc]キーでとめて、WAITコマンドをいれる

```
35 WAIT 3↵
```

さゆうキーでうごくように

```
36 X=X-BTN(28)+BTN(29)↵
```

あたりはんていをつけて、ゲームのできあがり！

```
40 IF SCR(X,5)=0 GOTO 20↵
```

あたっていないときだけ、くりかえし！

やってみよう！

1. はやさをちょうせいしよう
2. じぶんのキャラをかえてみよう（ALTおしながらA～Vをおす）
3. しょうがいぶつを"****"とふやしてみよう
4. ゲームのストーリーをかんがえてみよう
5. うらわざ（=バグ）をさがそう
6. バグをつぶすほうほうをかんがえよう

ジャンプゲーム

ジャンプでてきをよけるジャンプゲーム！

まずはしゅじんこう、とうじょう！

```
10 CLS:Y=15:V=0↵
20 LC 0,Y:?"M"↵
```

なにかキーをおしたら、ジャンプ！

```
30 IF Y=15 V=3*(INKEY()=0)↵
40 Y=Y-V:V=V-1↵
50 SCROLL LEFT↵
60 GOTO 20↵
```

てきキャラをだす

```
25 IF RND(10)=0 LC 31,15:?"K"↵
```

スピードをちょうせい

```
35 WAIT 4↵
```

ゲームオーバーはんていをくわえて、できあがり！

```
37 IF SCR(1,Y) END↵
```

やってみよう！

1. しゅじんこうをRにかえてみよう
2. しゅじんこうをALTおしながらRにかえてみよう
3. RND(10)をRND(4)にかえてみよう
4. てきキャラをすくなくしてみよう
5. スピードをちょうはやくしてみよう
6. $V=3*$ を $V=4*$ にかえてみよう
7. ジャンプ力をちいさくしよう
8. CLTとTICK()をつかって、スコアをついかしてみよう
9. ジャンプのかいすうを、スコアにしてみよう
10. かいぞうしてオリジナルゲームをつくろう

ジャンケンゲーム&おみくじ

グー、チョキ、パー、3つので、でたためにきめるにはランダム3

```
?RND(3)+
```

0をグー、1をチョキ、2をパーときめてみます

```
10 LET A,RND(3)+
20 IF A=0 ?"GU"+
30 IF A=1 ?"CHOKI"+
40 IF A=2 ?"PA"+
RUN+
```

RUNまたはF5キーをなんどもおしてみよう！

「じゃんけん」のかけごえのあと、じぶんのをいいながら、F5キー(RUN)でしようぶ！

```
20 IF A=0 ?"DAIKICHI"+
30 IF A=1 ?"KICHI"+
40 IF A=2 ?"KYO"+
```

LIST(F4キー)でひょうじする、20～40ぎょう、もじをかえれば「おみくじアプリ」に、はやがわり！

```
10 LET A,RND(4)+
50 IF A=3 ?"DAIKYO"+
```

ランダムのかずをふやして、A=3のぎょうをたせば、ひょうじパターンが変わります。

「かな」キーか「みぎAlt」キーをいちどおして、カタカナをローマじにゅうりょく。もういちどおすとアルファベットモードにもどるよ。

やってみよう！

1. ジャンケンにカタカナにしてみよう
2. ともだちのコンピューターどうしてジャンケンさせてみよう
3. あしたのてんきをうらなうツールをつくってみよう
4. コインのおもてかうらかアプリをつくってみよう
5. ガチャガチャをつくってみよう
6. レア（なかなか、あたらないもの）をつくってみよう

あっちむいてほい

コンピュータ(CPU)と「あっちむいてほい」するゲーム！

まずは、かけこえ「あっちむいて」

```
10 CLS➤  
20 ?"ACCHI MUI TE"➤
```

あなた(YOU)は、じょうげさゆうのカーソルキーで、むきをきめる。

```
30 A=INKEY()➤  
40 IF A<28 OR A>31 GOTO 30➤
```

※ひだり (←) は28、みぎ (→) は29、うえ (↑) は30、した (↓) は31

コンピュータ(CPU)のむきはランダムにきめる。

```
50 C=RND(4)+28➤
```

あなた(YOU)とコンピュータ(CPU)のむきと、けっかをひょうじ。

コンピュータ(CPU)とちがうむきなら、あなた(YOU)のかち。

おなじむきなら、あなた(YOU)のまけ。

```
60 ?"HOI!"➤  
70 ?"YOU:";CHR$(224+A-28)➤  
80 ?"CPU:";CHR$(224+C-28)➤  
90 IF A=C ?"LOST!" ELSE ?"WON!"➤
```

※キャラクターコード224：ひだり、225：みぎ、226：うえ、227：した

やってみよう！

1. まけるまでゲームがつづくようにしてみよう
2. たいせんするかずをきめてみよう
3. かったかずとまけたかずをかぞえてみよう
4. きめたじかんまでに、カーソルキーをおさなかったらまけにしてみよう
5. コンピュータどうしてたたかわせてみよう
6. じゃんけんとくみあわせてみよう

ねこにこばんゲーム - IchigoJam

ねこ、よこ10、たて5にとうじょう（キャラクターコード236、ALT+C）

```
10  CLS:X=10:Y=5↵
50  LC X,Y: ?CHR$(236)↵
```

カーソルでうごかせるようにする

```
20  K=INKEY()↵
30  X=X-(K=28)+(K=29)↵
40  Y=Y-(K=30)+(K=31)↵
60  WAIT10↵
70  LCX,Y:?"  "↵
80  GOTO20↵
```

こばんをばらまき、てんすう(S)つか

```
15  S=0:FOR I=1 TO 10:LOCATE RND(
30),RND(20): ?CHR$(245):NEXT↵
45  IF SCR(X,Y)=245 S=S+1:LC0,0: ?
S↵
```

10びょうでゲームオーバー

```
7   CLT↵
47  IF TICK()>600  END↵
```

やってみよう！

1. LISTをひょうじしてみよう
2. こばんをたいりょうにだしてみよう
3. 5びょうでゲームオーバーにしてみよう
4. はやくうごけるようにしてみよう
5. スコアが10てんずつあがるようにしてみよう
6. すきなようにかいぞうしてほぞんしよう

じんこうシミュレーション

にほんじんじんこうは、1おく2513まんにん。1ねんで27まんにんへりました。
(2016.11 そうむしょう、とうけいきよく、じんこうすいけい)

```
10 P=125364
20 D=274
```

Pにじんこう、Dにへるかずを「まんにん」をたんいとしてセット

2016ねんから、まいとしおなじようにへりつづけると、2100ねんのじんこうは？

```
30 FOR Y=2016 TO 21004
40 ?Y,P;4
50 ?4
60 P=P-D4
70 NEXT4
```

グラフにしてみよう（'はわりざんのきごう）

```
42 FOR I=0 TO P/10004
45 ?"*";:NEXT4
```

へりかたがまいとし2まんにんずつふえると？

```
55 D=D+24
```

やってみよう！

1. ESCキーでとめて、2060ねんのじんこうをかくにんしよう
2. へりかたをまいとし3まんにんにしてみよう
3. グラフのもじ"*"をべつのもじにかえてみよう
4. 42ぎょう、わっている1000を100にかえてみよう
5. じんこうをふやしてみよう
6. せかいじんこうは73おくにん、まいとし1おくにんふえると？

タッチタイピング

キーボードをみずにうつつタッチタイピングは、はやい！たのしい！かっこいい！

```
10 S="JKL"↵
20 CLT:FOR I=1 TO 10↵
30 N=RND(LEN(S)):C=ASC(S+N)↵
40 ?CHR$(C)↵
50 K=INKEY():IF K=0 CONT↵
60 IF C<>K ?"MISS!":END↵
70 NEXT: ?TICK()/60↵
```

みぎてのひとさしゆびを「J」、のいち、なかゆびを「K」、くすりゆびを「L」ランダムにひょうじされる、JかKかLをすばやくタイプ！
10かいまちがえずにうてるかな？もくひょうタイムは、8びょう！

クリアしたら、とうじょうするもじをだんだんふやそう。

ひだりてのひとさしゆびを「F」、FJKLの4しゅるいで8びょうだせるかな？

```
10 S="FJKL"↵
```

ひだりてのなかゆび「D」、くすりゆび「S」、こゆび「A」
これがタッチタイピングのホームポジション！

```
10 S="ASDFJKL"↵
```

やってみよう！

1. ASDFJKLできろく8びょうをだそう
2. みぎてのひとさしゆびでおす「H」をくわえよう
3. ひだりてのひとさしゆびでおす「G」をくわえよう
4. したのだん「ZXCVBNM」をくわえよう
5. うえのだん「QWERTYUIOP」をくわえよう
6. アルファベットぜんぶできろく8びょうをだしてみよう

Q	W	E	R	T	Y	U	I	O	P
A	S	D	F	G	H	J	K	L	;
Z	X	C	V	B	N	M	,	.	/

かくゆびのキーたんとうひょう

ナインパズル - IchigoJam

1~8のパネルをスライドさせてきれいにならべるパズル

```
10 CLS:FOR I=0 TO 7:LC I%3,I/3:?  
I+1:NEXT+
```

0から7までのくりかえしで、かんせいパターンをひょうじ
(%はあまりをけいさんしてくれます)

パネルのスライドはカーソルで

```
20 X=2:Y=2+  
30 LC X,Y:?"*"+  
40 K=INKEY():GOSUB 60+  
50 GOTO 30+  
60 LC X,Y+  
70 X=X-(K=28)*(X>0)+(K=29)*(X<2)  
:Y=Y-(K=30)*(Y>0)+(K=31)*(Y<2)+  
80 ?CHR$(SCR(X,Y)):RETURN+
```

(GOSUBはRETURNで「もどる」きのうがついたGOTOです)

ぐちゃぐちゃに200かいかきませよう

```
25 FOR I=1 TO 200:K=RND(4)+28:GO  
SUB 60:NEXT:CLT+
```

(RNDでキーのがちゃがちゃをさいげん!)

クリアしたかどうかのはんていと、クリアタイムをびょうでひょうじ!

```
35 F=0:FOR I=0 TO 7:F=F+(SCR(I%3  
,I/3)=I+49):NEXT:IF F=8 LC0,4:?"  
CLEAR! TIME:";TICK()/60;"s":PLAY  
"05E16E":END+
```

やってみよう!

1. かきませるかいうすをかえてみよう
2. パズルゲームスタートのおとをついかしよう
3. うごかしたときにこうかおんをならそう
4. クリアしたときにLEDをピカピカひからせよう
5. さいこうきろくのきおくとをひょうじをついかしよう

ガチャシミュレーション - IchigoJam

アルファベットをガチャでコンプリートしよう！

```
10 A=RND(26)*#  
20 ?"DRAW:";CHR$(ASC("A")+A)*
```

アルファベットはAからZまで26しゅるい。RNDをつかってガチャ！

コレクションをひょうじしよう

```
30 [A]=[A]+1*  
40 FOR I=0 TO 25*  
50 LC I%6*5,5+I/6*3*  
60 ?CHR$(ASC("A")+I);": ";[I]*  
70 NEXT*
```

コンプリートするまで、じどうガチャ！

```
5 CLS:CLV*  
35 N=N+1:C=0*  
65 IF [I] C=C+1*  
80 ??:?:?"COLLECT:";C;"/26 ";*  
90 ?N;"00YEN"*  
100 IF C=26 ?"COMPLETE!":END*  
110 GOTO 10*
```

いくらかかるかな？

Aをレアにしてみよう

```
15 IF A=0 IF RND(10)>0 A=1*
```

Aがでても10ぶんの9のかくりつでBにしちゃう

やってみよう！

1. 3かいためして、くらべてみよう
2. AからFまでをレアにしてみよう
3. Zがでないようにしてみよう
4. 2ぶんの1のかくりつでZをだしてみよう
5. かぶらないようにしてみよう

九九ゲーム - IchigoJam

九九のひみつにせまってみよう！

```
10 CLS: INPUT "DAN(1-9): ", D↵
20 FOR I=1 TO 10↵
30   LC 2,18: ?D; "*" ; I ; "=" ; ↵
40   INPUT " ", N↵
50   IF N<>I*D LC 2,20: ?D; "*" ; I ; "
=" ; I*D: END↵
60 NEXT↵
```

うちこんだら RUN(またはF5)、だんをいれてエンター、そのだんの九九クイズが
でるよ。すうじとエンターキーでこたえよう！

ヒントひょうじきのうをついか！

```
25 GOSUB 100↵
70 END↵
100 CLS↵
110 FOR J=1 TO I↵
120   FOR K=1 TO D↵
130     LC J*2,12-K: ?CHR$(237)↵
140   NEXT↵
150   LC J*2,13: ?D; ↵
160   IF J<I ?" + "↵
170 NEXT↵
180 RETURN↵
```

やってみよう！

1. 20行の10を9にかえると、かける9までになるよ
2. 20行の1を2にかえると、かける1をとばせるよ
3. 10行を 10 D=RND(9)+1 にかえるとどのだんがでるかわからないゲームになるよ
4. 130行の237を255にかえるとネコがイチゴにかわるよ
5. 15 CLT と 65 ?TICK()/60 をくわえるとクリアするまでのタイムがでるよ
6. 九九をつかったおもしろゲームをかんがえてみよう！

すうじタイピングゲーム

なまえタイピングゲーム（20ぎょうをじぶんのなまえでうちこもう！）

```
10 CLS#  
20 ?"TAISUKE FUKUNO"#  
30 CLT:P=0#  
40 C=SCR(P,0)#  
50 IF C=0 ?:"TICK()/60:END#  
60 IF INKEY()<>C CONT#  
70 ?CHR$(C);:P=P+1#  
80 GOTO 40#
```

なれてきた？

ABCアルファベットタイピングゲームにかいぞう！

```
10 FOR I=1 TO 10: ?CHR$(65+RND(3))  
;:NEXT#
```

20びょうきれるかな？

すうじタイピングゲームにかいぞう！

```
20 FOR I=2 TO 5: ?RND(256);", ";:N  
EXT: ?RND(256)#
```

10びょうきれるかな？

スコアをしょうすうだい1いまでひょうじするには？

```
T=TICK(): ?T/60; ". "; T%60/6#
```

コンマ1びょうたんいで、しょうぶ！

やってみよう！

1. すきなことばタイピングゲームにしてみよう
2. むずかしいえいたんごをタイピングゲームでおぼえちゃおう
3. 2しんすうタイピングゲームにしてみよう
4. でたらめアルファベットタイピングゲームにしてみよう
5. えもじタイピングゲームにしてみよう

なんようびうまれ？

ねんなんがつなにちは、なんようび？じつはプログラムでけいさんできます。
まずは、ようびをひょうじするプログラム

```
10 LET[0],"SUN","MON","TUE","WED",  
  "THU","FRI","SAT"  
RUN  
?STR$([0])
```

STR\$をつかって、はいれつにいたたもじれつのせんとう[0]をひょうじします。

にちようびからどようびまでをくりかえす、7つのようび

```
FOR I=0 TO 22:STR$([I%7]):NEXT
```

INPUTをつかって、ねんをY、つきをM、ひをDにいれよう

```
20 INPUT"Y?",Y  
30 INPUT"M?",M  
40 INPUT"D?",D  
RUN
```

にゆうりよくされた、ねんがっぴをひょうじしてみよう

```
?Y,M,D
```

ねんがっぴから、ようびをけいさんする「ツェラーのこうしき」

```
50 IF M<3 M=M+12:Y=Y-1  
60 H=(Y+Y/4-Y/100+Y/400+(13*M+8)/5+D)/7  
70 ?STR$([H])
```

これでようびがわかっちゃうのはスゴイですね！

やってみよう！

1. きょうのひづけをいれて、ようびをかくにんしよう
2. じぶんのたんじょうびは、なんようびだろう？
3. つぎのたんじょうびは、なにようびだろう？
4. ようびをカタカナでひょうじしてみよう
5. たんじょうづきのカレンダーをつくってみよう

ひみつのあんごうをつくってみよう！

もじのしょうたいがわかったら もじをべつのもじにかえてあそんでみよう

```
10 S="IchigoJam"  
40 ?STR$(S)
```

RUNでじっこうしても、もとの IchigoJam がひょうじされるだけだね。

「もじコード」をいじて、べつのもじにかえちゃおう。

```
20 K=1  
30 FOR I=0 TO LEN(S)-1: ?CHR$(ASC(S+I)+K)::NEXT: ?
```

JdijhpKbn とひょうじされたかな？

IchigoJam が JdijhpKbn にあんごうかされたね。

20ぎょうめの K のすうじをかえるとちがうあんごうになるよ。

このすうじがあんごうの「かぎ」だね。ひみつにしておこう。

```
20 K=150
```

なぞのあんごうになったかな？

あんごうかされたもじをもとにもどしてみよう。

```
10 S="JdijhpKbn"  
20 K=1  
30 FOR I=0 TO LEN(S)-1: ?CHR$(ASC(S+I)-K)::NEXT: ?
```

IchigoJam にもどったかな？

30ぎょうめの K のけいさんをべつのものにかえるとどうなるかな？

```
30 FOR I=0 TO LEN(S)-1: ?CHR$(ASC(S+I)^K)::NEXT: ?
```

やってみよう！

1. もとにもどすプログラムをともだちにわたしてあんごうでおはなししてみよう
2. うまくもとにもどらないことがあるよね。どうしたらいいかな？
3. あんごうかと もとにもどすプログラムをおなじプログラムにすることはできるかな？

さいきょうのあんごうワнтаイムパッド

かぎをいちどしかつかわない「あんごう」ワнтаイムパッドは、げんりてきにやぶることがふかのうらしい。

```
10 LET[0],8,30,100,38,10,20,5
```

でたらめなかずを、つたえたいもじすうぶんつくって、たいせつなひとにかぎをこっそりわたしておこう。（さいだい102もじ）

ひともじずつキーをおして、あんごうかされたかずをメモしよう。

```
30 FOR I=0 TO 101
40 K=INKEY():IF K=0 CONT
50 A=K^[I]
60 ?CHR$(K); "<"; CHR$(A), A
70 NEXT
```

きごう「^」はXOR（はいたてきろんりわ）、2かいXORするともとどおりになる。

ふくごう（あんごうをもとにもどすこと）プログラムは、かずをいれて、もじをひょうじ。

```
100 FOR I=0 TO 101
110 INPUT N
120 A=N^[I]
130 ?CHR$(A), A
140 NEXT
```

あんごうかされたメモに、かかれたかずをひとつずついれてみよう

やってみよう！

1. かぎのながさいじょうにキーをおすとどうなるかためしてみよう
2. XOR「^」を2かいするともとにもどることをかくにんしてみよう
3. BIN\$をつかってXORのなぞにせまってみよう
4. かぎなしに、メモからふくごうするほうほうをかんがえてみよう
5. かぎがだれかにばれたらどうなるかそうぞうしてみよう
6. かぎをつかいまわしたらどうなるかそうぞうしてみよう

あんごうのかぎをあいてにわたすには？

あんぜんにあんごうをつかうには「かぎ」はほかのひとにしられてはいけないよね。「かぎ」そのものをあいてにつたえなくても、おなじ「かぎ」をつかえるとかべつなけいさんをプログラミングでつくってみよう！

まずはとかべつなすうじをじゅんびしよう。ひとつえらんでつかうよ。

```
10 LET[0],53,59,61,67,83
20 FOR I=0 TO 4:?[I];" ";NEXT: ?
30 INPUT"P=? ",P
```

つぎはじぶんだけのひみつのすうじをにゅうりよく。

```
40 INPUT"スウジ? ",A
```

かぎのあたいをしょきかして、けいさんをじっこう！

```
50 K=1
60 GSB@CALC
```

けいさんするぶぶんをつくるよ。

```
500 @CALC
510 FOR I=0 TO A
520   K=K*2
530   K=K%P
540 NEXT
550 RTN
```

けいさんのけっかをあいてにつたえよう。

このすうじはほかのひとにしられてもだいじょうぶだよ。

```
70 ?"ワタススウジ? ";K
```

おなじプログラムでけいさんした、あいてからうけとったすうじをいれる。

このときのあいてとは、おなじPをつかってね。

```
100 INPUT"ウケトッタスウジ? ",K
110 GSB@CALC
120 ?"カギ"スウジ? ";K
130 END
```

でてきたすうじはあいてとおなじになっているよね！

このすうじを「かぎ」としてつかおう！

やってみよう！

1. ひみつのすうじをいろいろとかえてためしてみよう

2. けいさんぶぶんの % というけいさんはどんなけいさんかな？しらべてみよう
3. とくべつなすうじは「そすう」というよ。どんなとくちょうがあるかな？

CC BY IchigoJamプリント <http://ichigojam.net/print/IchigoJam@jig.jp>

FLOPPY BIRD - IchigoJam

FLOPPY BIRDを作ってみよう

```
10 X=3↵
20 Y=3↵
30 H=0↵
40 W=31↵
50 I=0↵
60 J=0↵
70 CLS↵
80 H=RND(14)+3↵
90 IF J%10<>0 GOTO 150↵
100 LC W,I: ? "I"↵
110 IF H=I I=I+5↵
120 I=I+1↵
130 IF I<22 GOTO 90↵
140 I=0↵
150 Y=Y+1-BTN(UP)*2↵
160 IF Y>23 : GOTO 200↵
170 IF Y<0 : Y=0↵
180 SCROLL LEFT↵
190 IF SCR(X,Y)=0 GOTO 220↵
200 PRINT "GAME OVER" : WAIT 60↵
210 END↵
220 LOCATE X,Y: PRINT "*"↵
230 WAIT 6↵
240 LOCATE X,Y: PRINT " "↵
250 J=J+1↵
260 GOTO 80↵
```

*がじぶんだよ うえのキーをおすとジャンプするよ

みぎからくるかべをうまくよけよう！

やってみよう！

1. じぶんのすがたやかべをかえてみよう
2. かべのはやさをかえてみよう
3. かべのあなのおおきさをかえてみよう

明日天気になあれ

天気を占ってみよう！F5でランダムに表示されます！

```
10 LET A,RND(4)
```

ランダムに4つ（晴、雨、曇、雪）に決めるのでランダム4とおこう！

次に、0を晴、1を雨、2を曇、3を雪に指定してみよう！

```
20 IF A=0 ?"HARE"  
30 IF A=1 ?"AME"  
40 IF A=2 ?"KUMORI"  
50 IF A=3 ?"YUKI"
```

F5で占ってみよう！

やってみよう！

1. 雪を台風に変えてみよう！
2. 天気の種類を増やしてみよう！（RNDや60）
3. 工夫しておみくじやラッキーカラー占いを作ってみよう！

CC BY [IchigoJamプリント](http://ichigojam.net/print/) <http://ichigojam.net/print/> IchigoJam@jig.jp

思いを伝えてみよう！

気持ちを乗せて、書いていく！

```
10 LC 10,9:?"LOVE LETTER"
```

文のタイトルを書こう

最初の10で書いたコードの動く順番を決める

LCの後の10,9で位置決め

その後にタイトルを書く

```
20 IF INKEY()=0 CONT
```

ボタンを押したら次に進むようにしてあげる

ページの最後にこのコードをいれることでボタンで進むようになるよ

```
30 LC 10,10:?"ICHIGOJAM SANHE"
```

伝えたい内容を考えて、書いてみよう！

どんどん書いていくよ！

```
40 LC 10,10:?"ZUTTO MAE KARA"
```

最後に

1. コードが動かしたい順番になってるか確認！
2. 最後のページは、もちろんあの言葉！
3. 出来上がったら焦らずに、RUNを押して確認だ！
4. 最終確認が終わったらIchigoDakeを渡そう！

数字の滝でハッカー気分

何かを表示させるには、「PRINT」を使う

```
1 PRINT 1
```

もっと数字が欲しいので、もう一行かいていこう

```
1 PRINT 1
2 PRINT 2
```

しかし、こう書いてしまうとすぐに2つめが表示されてしまう。滝のように流れてるとはいえない。と、いうわけでちょっと待ってもらおう。

「WAIT」コマンドを使うと次の行に行くのを待ってくれる。
今回は100にしよう。（WAIT 60で1秒） 少し遅れて表示される。

```
1 PRINT 1
2 WAIT 100
3 PRINT 2
```

どんどん行を増やしていこう。「PRINT 10」までやってみよう。「WAIT」を100から10ずつへらしていこう。

```
4 WAIT 90
5 PRINT 3
6 WAIT 80
7 PRINT 4
8 WAIT 70
9 PRINT 5
10 WAIT 60
11 PRINT 6
12 WAIT 50
13 PRINT 7
14 WAIT 40
15 PRINT 8
16 WAIT 30
17 PRINT 9
18 WAIT 20
19 PRINT 10
20 WAIT 10
```

PRINT10まで書いてくともう早くしようがない。

「GOTO」を使って繰り返しにし、さらに数字をランダムに表示してみよう

```
21 PRINT RND(99)+1
22 GOTO 21
```

やってみよう！

1. なぜ、RNDに+1をするのか説明してみよう

2. WAITの時間を自分好みにしてみよう
3. 他のコマンドでもっとスマートに書いてみよう

CC BY IchigoJamプリント <http://ichigojam.net/print/> IchigoJam@jig.jp

たいせん、ばとろわゲーム

ふたりたいせんゲームのプレイヤーをひょうじしよう

```
10  CLS : X=5 : Y=10 : V=1 : W=0 : A=26 : B=1
0 : C=-1 : D=0 : F=0
20  LC  X,Y : ? "1" : LC  A,B : ? "2"
```

V,W,C,Dはすすむほうこう、Fはフラグ

```
30  X=X+V : Y=Y+W : A=A+C : B=B+D
40  WAIT 3
50  GOTO 20
```

ひだりプレイヤーはWASD、みぎプレイヤーはカーソルキーでそうさ

```
21  K=INKEY( )
22  IF  K=65  V=-1 : W=0
23  IF  K=68  V=1 : W=0
24  IF  K=87  V=0 : W=-1
25  IF  K=83  V=0 : W=1
26  IF  K=28  C=-1 : D=0
27  IF  K=29  C=1 : D=0
28  IF  K=30  C=0 : D=-1
29  IF  K=31  C=0 : D=1
```

じょうがいはんていと、あたりはんていをつけよう

```
45  IF  X<0 OR X>31 OR Y<0 OR Y>22
OR  SCR(X,Y)  F=F+2
46  IF  A<0 OR A>31 OR B<0 OR B>22
OR  SCR(A,B)  F=F+1
50  IF  F=0 GOTO 20
```

しょうしゃをたたえよう

```
60  IF  F=1 ? "P1 OME!"
70  IF  F=2 ? "P2 OME!"
80  IF  F=3 ? "DRAW..."
```

やってみよう！

1. キャラや、はやさをかえてみよう！
2. メッセージをかえてみよう！

